
Kapitel 1

Algorithmusbegriff und algorithmische
Sprachen

1.1 Algorithmen und ihre Darstellung

Wesentliche Voraussetzungen für eine erfolgreiche Problemlösung:

– eine exakte Aufgabenstellung

– eine Algorithmierung des Lösungsweges

– eine fehlerfreie Formulierung des Lösungsverfahrens als Programm

1.1.1 Anschaulicher Algorithmusbegriff

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

1. Vorbetrachtung:
x1,2 = (−a

2
) ±

√

(−a
2
)2 − b

Falls D = (−a
2
)2 − b < 0 , so sind x1 und x2 konjugiert

komplex mit x1,2 = Re ± Im ∗ i .
2. Algorithmierung:

– Gegebene Größen (Eingabegrößen)

– Gesuchte Größen (Ausgabegrößen, Lösungen)

– Umformungsregeln (Regelsystem)

Algorithmus in Schrittform:

S1 Gib a und b ein!

S2 Setze c = −a/2
S3 Setze D = c ∗ c− b

S4 Setze w =
√

|D|
S5 Falls D < 0 ist, so gehe zu S9, sonst zu S6

S6 Berechne x1 = c+ w , x2 = c− w

S7 Gib x1 und x2 und “Lösungen reell!“ aus

S8 Gehe zu S11

S9 Setze Re = c , Im = w

S10 Gib Re, Im und “Lösungen komplex!“ aus

S11 STOP

Definition 1 : Intuitiver Algorithmusbegriff

Sei P eine Problemklasse und A die Menge der Daten, die P beschrei-

ben. Man bezeichnet mit

A – Menge der Eingabegrößen (Problemdaten, Eingabeinformationen)

B – Menge der Ausgabegrößen (Lösungsmenge, Ausgabeinformationen).

Ein Algorithmus AP : A → B
• ist ein determiniertes, schrittweises Verfahren,

• das für jedes a ∈ A nach endlich vielen Schritten anhält und

• eine Lösung b ∈ B liefert.

Eine Funktion f : A → B heißt berechenbar, Falls ein Algorithmus

AP mit AP (w) = f(w) ∀w ∈ A existiert.



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 3

Darstellungsformen für Algorithmen:

– Schrittalgorithmus

– Programmablaufplan (PAP)

– Struktogramm

– Computerprogramm

Schrittalgorithmus

Der Schrittalgorithmus ist eine 1-dimensionale Form des Algorithmus mit

verbaler rechnerunabhängiger Darstellung der Algorithmusschritte.

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

S1 Gib a und b ein!

S2 Setze c = −a/2
S3 Setze D = c ∗ c− b

S4 Setze w =
√

|D|
S5 Falls D < 0 ist, so gehe zu S9, sonst zu S6

S6 Berechne x1 = c+ w , x2 = c− w

S7 Gib x1 und x2 und “Lösungen reell!“ aus

S8 Gehe zu S11

S9 Setze Re = c , Im = w

S10 Gib Re, Im und “Lösungen komplex!“ aus

S11 STOP

Programmablaufplan (PAP)

Der Programmablaufplan ist eine weitgehend rechnerunabhängige 2-

dimensionale grafische Algorithmusdarstellung, bei der die Schritte durch

Kästen definiert und durch Linien miteinander verbunden werden.

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

S t a r t

E:  a, b

w := sqrt(abs(D))

c  :=  - a / 2
D :=  c*c - b

D  <  0  ?ja nein

Re  :=  c
Im  :=  w

x1  :=  c + w
x2  :=  c - w

A:  Re, Im,
"Lsg.  komplex"

 A:  x1, x2, 
"Lsg.  reell"

 S t o p p

Struktogramm

Das Struktogramm ist eine weitgehend rechnerunabhängige 2-dimensio-

nale grafische Algorithmusdarstellung innerhalb eines fest vorgegebenen

Rahmens. In diesen Rahmen können beliebig komplizierte Strukturblöcke

“kantendeckend“ eingefügt werden.

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

Deklarationen

reell :  a, b, c, D, w, x1, x2, Re, Im

 Anweisungen

E :  a, b

c   :=  - a/2
D  :=  c*c - b
w  :=  sqrt(abs(D))

D  <  0 ?
ja  nein

Re  :=  c

Im  :=  w

x1  :=  c + w

x2  :=  c - w

A :  Re, Im, 

"Lösung komplex"

A :  x1, x2, 

"Lösung reell"

(Computer)-Programm

• Ein Programm ist eine geordnete oder strukturierte Menge von An-

weisungen zur Abarbeitung eines Algorithmus.

• Ein Computerprogramm ist die Formulierung eines Algorithmus in

maschinell abarbeitbarer Form.

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

Program Quadratische_Gleichung;

var

a, b, c, D, w, x1, x2, re, im : real;

begin

readln (a, b);

c := -a / 2.0;

D := c * c - b;

w := sqrt (abs (D));

if D < 0 then

begin

re := c; im := w;

writeln(re, im, ’Loesungen komplex’)

end

else

begin

x1 := c + w; x2 := c - w;

writeln (x1, x2, ’Loesungen reell’)

end

end.

1.1.2 Exakter Algorithmusbegriff

Frage: Existiert eine (technische) Aufgabe, die ein Computer jetzt und

zukünftig nicht erfüllen kann – d.h irgendeine Aufgabe, für die kein Al-

gorithmus existiert?

Geschichte der Berechenbarkeit

• Glaube der klassischen Mathematik:

“Wenn irgendein Problem präzise dargelegt werden kann, so gibt es

am Ende stets eine Lösung oder es kann bewiesen werden, daß keine

Lösung existiert.“

• Gottfried Wilhelm Leibniz (1646-1716):

“Es ist ein Algorithmus zu konstruieren, der es gestattet, jede be-

liebige mathematische Aufgabe zu lösen.“

• David Hilbert (1862-1943): Entscheidungsproblem

“Zu finden ist ein Algorithmus, der zu einer beliebigen gegebenen

Aussage in einem Axiomensystem entscheiden kann, ob die Aussage

wahr oder falsch ist.“

• Kurt Gödel (1906-78): Beweis des Unvollständigkeitssatzes, 1931

“Es gibt keinen Algorithmus, der als Eingabe irgendeine Aussage

über die natürlichen Zahlen enthält, und dessen Ausgabe feststellt,

ob diese Aussage wahr oder falsch ist.“

• Entscheidungsproblem der mathematischen Logik:

“Von je 2 Formeln R und S eines Logikkalküls ist zu entscheiden,

ob eine deduktive Kette existiert, die von R nach S führt oder

nicht.“

Satz von Church, 1936 : “Das Entscheidungsproblem der mathema-

tischen Logik ist algorithmisch unlösbar.“

• Algorithmische Unlösbarkeit von Problemen der Informatik:

1. Halteproblem: Es existiert kein AlgorithmusA, der zu einem be-

liebigen Programm P und beliebigen Eingabedaten D mitteilt,

ob P anhält oder eine Endlosschleife enthält.

2. Äquivalenzproblem: Es existiert kein Algorithmus A, der zu 2

beliebigen Programmen P1 und P2 feststellen kann, ob sie die

gleiche Aufgabe erfüllen, d.h. bei gleicher Eingabe stets dieselbe

Ausgabe erzeugen.

• Algorithmusdefinitionen :

– Kurt Gödel: Algorithmus als Folge rekursiver Funktionen

– Alonso Church: Lambda-Kalkül

– Alan Turing: Algorithmus als Funktionsschema einer hypotheti-

schen Maschine

Beweis der Äquivalenz dieser Algorithmusdefinitionen ⇒ eine Defi-

nition ist hinreichend!

• Church–Turing–These :

“Alle vernünftigen Definitionen von Algorithmus sind gleichwertig

und gleichbedeutend.“

Die Turingmaschine, 1936

(Alan M.Turing,1912-54)

1. Aufbau der Turingmaschine

Der Mensch besitzt... Die Turingmaschine besitzt...

1 Alphabet (mit Zif- Bandalphabet B

fern u. Sonderz.) (verarbeitbare Zeichen)

2 Bogen Papier Rechenband mit

(äußerer Speicher) Eingabealphabet E ⊂ B

mit geg. Problem (Symbole auf dem Band)

3 Gedächtnis Innere Zustände

(innerer Speicher) (Zustandsmenge S )

4 Auge und Lesevorrichtung (LV)

Bleistift Schreibvorrichtung (SV)

5 Steuerung von Auge Verschiebevorrichtung (VV)

und Bleistift (mit 3 Verschiebungen aus V)

6 Fähigkeit zu ele- Überführungsfunktion δ mit:

mentaren Rechen- – Zustandsfunktion

operationen – Ausgabefunktion

– Verschiebefunktion

E = {e1, e2, ..., er} Eingabealphabet (geg. Symbole)

B = {b1, b2, ..., bm} Bandalphabet mit E ⊂ B

S = {s1, s2, ..., sn} Zustandsmenge

V = {L,M,R} Menge der 3 Verschiebungen

Das Spezialsymbol ∗ kennzeichnet oft ein Feld als leer.

2. Funktionsschema der Turingmaschine (Turing-Tafel)

Beispiel S = {s1, s2, s3, s4, s5} , B = {λ,♥, α, β}

S
s1 s2 s3 s4 s5

B

λ λRs4 λLs3 λRs1 λMs5 λMs5

♥ αMs2 βMs1 ♥Rs1 ♥Ls1 ♥Ms5

α αLs1 αRs2 ♥Ls3 λRs4 αMs5

β βLs1 βRs2 λLs3 ♥Rs4 βMs5

Anfangszustand (0. Konfiguration):

♥ ♥ ♥ ♥ ♥ ♥ ♥ ♥ ♥ ♥
s1

1. Konfiguration:

♥ ♥ ♥ α ♥ ♥ ♥ ♥ ♥ ♥
s2

2. Konfiguration:

♥ ♥ ♥ α ♥ ♥ ♥ ♥ ♥ ♥
s2

3. Konfiguration:

♥ ♥ ♥ α β ♥ ♥ ♥ ♥ ♥
s1

usw. usw.

“Letzte“ Konfiguration:

♥ ♥
s5

EUKLIDischer Algorithmus – vereinfachte Darstellung

der Turing-Tafel

S
s1 s2 s3 s4

B

λ Rs4 Ls3 Rs1 !

♥ αs2 βs1 Rs1 Ls1

α L R ♥L λR

β L R λL ♥R

Definition 2 : Turing–Maschine

1. T = (E,B, S, δ, s0, F ) ist eine Turing-Maschine, wenn

• die 3 nichtleeren Mengen

E = {e1, e2, ..., er} Eingabealphabet

B = {b1, b2, ..., bm} Bandalphabet mit E ⊂ B

S = {s1, s2, ..., sn} Zustandsmenge

• die 2 Zustände

s0 ∈ S Anfangszustand

F ⊂ S Menge der möglichen Endzustände

• die Überführungsfunktion δ : S ×B −→ S × V ×B

definiert sind.

2. Existiert zu einer Funktion f(x1, x2, ..., xn) eine Turing-Maschine,

die nach endlich vielen Schritten anhält, so heißt sie Turing-berechenbar.

3. Church-These: Die Klasse der intuitiv berechenbaren Funktionen ist

gleich der Klasse der Turing-berechenbaren Funktionen.

1.2 Algorithmische Sprachen

Die Notwendigkeit der Betrachtung (formaler) Sprachen resultiert aus

dem Bedürfnis, Mitteilungen an den Computer zu übergeben. Mitteilun-

gen komplizierter Art sind jedoch an eine Sprache gebunden.

Formale Sprachen (z.B. Programmiersprachen)

• Syntaktische Definition der Sprache: Festlegung darüber, welche

Zeichenkette ein Programm der Sprache ist und welche nicht!

• Notwendigkeit der exakten formalisierten Beschreibung:

– Programmierer und Computer müssen die gleichen Programme

als richtig bzw. falsch ansehen!

– Programme sollen auf möglichst vielen Computern unter möglichst

vielen Betriebssystemen laufen!

– Programme sollen der Dokumentation von Algorithmen dienen!

1.2.1 Generative Grammatiken

Elemente formaler Sprachen

• Alphabet T : Menge aller Grundsymbole (Terminalsymbole, ter-

minals) der Sprache

• Sätze ϕ der Sprache: Aneinanderreihung von Zeichen zu Zeichen-

ketten (Sätzen) der Sprache:

ϕ = z1z2z3...zn , zi ∈ T

Länge |ϕ| der Zeichenkette = Anzahl der Zeichen von ϕ

• Zeichenkettenmenge T ∗ : Menge aller möglichen Zeichenketten

endlicher Länge über dem Alphabet T

• Leere Zeichenkette ε :

Sie gehört zu jeder Zeichenkettenmenge T ∗ und hat die Länge 0.

• Formale Sprache L :

Eine formale Sprache ist eine Teilmenge L ⊂ T ∗ .

• Syntax P der Sprache L:

Die Regeln, die entscheiden, welche Elemente aus T ∗ zur Sprache

L gehören, bilden die Syntax von L. Sie beschreiben, auf welche

Weise die Zeichen aus T zu kombinieren sind, um zu einem Satz

ϕ der Sprache L zu gelangen (“Produktionsregeln“).

• Semantik der Sprache L:

Sie beschreibt die Bedeutung der Zeichen bzw. Zeichenkombinatio-

nen, d.h. sie ist ein Regelwerk zur inhaltlichen Gestaltung der Sätze.

Beispiel 1 Alphabet sei T = {a,+}

1. Zeichenkettenmenge

T ∗ = {ε, a,+, aa, a+,+a,++, aa+, ...}

2. Syntax einer Sprache SUMME

“Zur Sprache SUMME gehören alle die Sätze, die aus einem

A, gefolgt von n (n ≥ 0) Zeichenketten +a bestehen.“

Damit ist

SUMME = {a, a+ a, a+a+a, a+ a+ a+ a, ...} ⊂ T ∗

3. Semantik der Sprache SUMME:

a ist als Variable und + als Additionsoperator zu interpretieren.

Nachteil: Explizite Aufzählung aller Sätze der Sprache ist i.allg. nicht

möglich!

Generative Grammatiken G

• Bedeutung:

Sie liefern eine Konstruktionsvorschrift zur Erzeugung derjenigen

Zeichenketten aus T ∗ , die zu einer Sprache L mit dem Alpha-

bet T gehören =⇒ prinzipiell sind sämtliche Sätze der Sprache

angebbar.

• Metasymbole:

Das Alphabet T der Grundsymbole (terminals) wird durch Meta-

symbole (non-terminals) zu einem Vokabular V erweitert. Dann ist

V − T die Menge der Metasymbole.

• Syntaktische Regeln P (Produktionen):

Sie werden als geordnetes Paar (u, v) ∈ P dargestellt. Dabei ist

u ∈ V ∗ − T ∗ und v ∈ V ∗ − {ε}

Wirkung: Stehen u und v in der Relation (u, v), so darf u in jeder

Zeichenkette durch v substituiert werden.

Definition 3 : Erzeugung (Ableitung)

(i) Für ϕ,ψ ∈ V ∗ heißt ϕ → ψ einfache Erzeugung (Ableitung)

von ψ aus ϕ, wenn es Zeichenketten ϕ1, ϕ2, u, v ∈ V ∗ gibt, so

daß gilt

ϕ = ϕ1uϕ2 , ψ = ϕ1vϕ2 mit (u, v) ∈ P .

(ii) Für ϕ,ψ ∈ V ∗ heißt ϕ
∗→ ψ eine Erzeugung (Ableitungsfolge)

von ψ aus ϕ, wenn es Zeichenketten ϕi ∈ V ∗ für

i = 0, 1, 2, ..., n gibt, so daß gilt

ϕ0 = ϕ, ϕn = ψ und ϕi → ϕi+1 mit i = 0(1)n−1 .

Beispiel 2

• Alphabet: T = {a, b, c}
• Vokabular: V = {A,B,C,D} ∪ T
• Syntaktische Regeln:

P = {(A, aD), (A, aAD), (D,BC), (CB,BC),

(aB, ab), (bB, bb), (bC, bc), (cC, cc)}
Beispiele für Erzeugungen:

aBB
(5)→ abB

(6)→ abb

A
(1)→ aD

(3)→ aBC
(5)→ abC

(7)→ abc .

Ursprungsketten sind aBB und A .

Definition 4 : Grammatik (A.N.Chomsky, 1959)

Eine Grammatik (des Types 0) ist ein Quadrupel

G = (V, T, σ, P ) mit

• V endliche nichtleere Menge (Vokabular)

• T echte nichtleere Teilmenge von V (Alphabet, Grundsymbole)

• σ ein Element aus V − T (Satzsymbol, Startsymbol)

• P endliche nichtleere Teilmenge von (V ∗ −T ∗) × (V ∗ − {ε})

(Produktionsregeln).

V − T sind die Metasymbole (non-terminals), T die Grundsymbo-

le (terminals). Die Ursprungskette ist auf ein einziges Metasymbol σ

reduziert!

Beispiel 2 (s.o.) G = (V, T, σ, P ) mit

• Alphabet T = {a, b, c}
• Vokabular V = {A,B,C,D, a, b, c}
• Satzsymbol σ = A

• Syntaktische Regeln

P = {(A, aD), (A, aAD), (D,BC), (CB,BC),

(aB, ab), (bB, bb), (bC, bc), (cC, cc)}

Welche Ableitungen aus dem Satzsymbol A sind möglich?

A
(∗)→ abc , A

(∗)→ aabbcc , A
(∗)→ aaabbbccc

bzw. allgemein ist die Menge aller aus A ableitbaren Zeichenketten

(Sätze)

L(G) = {x | x = anbncn , n ∈ N}

Definition 5 : Sprache

Eine durch eine Grammatik G = (V, T, σ, P ) definierte (formale)

Sprache L(G) ist die Menge

L(G) = {x | x ∈ T ∗ mit σ
(∗)→ x } .

Eine Sprache ist also die Menge aller Zeichenketten aus T ∗ , die aus

dem Satzsymbol (Startsymbol) σ erzeugbar (ableitbar) sind!

Beispiel 3

Sprache “Natuerliche Zahl“ G = (V, T, σ, P ) mit

• T = {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}
• V = {〈Ziffer〉, 〈Natuerliche Zahl〉} ∪ T
• σ = 〈Natuerliche Zahl〉
• Syntaktische Regeln:

P = {(〈Ziffer〉, 0), (〈Ziffer〉, 1), (〈Ziffer〉, 2), ..., (〈Ziffer〉, 9),

(〈Natuerliche Zahl〉, 〈Ziffer〉),
(〈Natuerliche Zahl〉, 〈Natuerliche Zahl〉〈Ziffer〉)}

Einige Erzeugungen aus dem Satzsymbol:

σ → 〈Ziffer〉 → 1 ,

σ → 〈Natuerliche Zahl〉〈Ziffer〉 → 〈Ziffer〉〈Ziffer〉 → 27 ,

σ
(∗)→ 4711 ,

σ
(∗)→ 007 ,

σ
(∗)→ 0 usw.

Definition 6 : Kontextfreie Grammatik

Eine Grammatik G = (V, T, σ, P ) heißt kontextfrei (oder: Typ-2-

Grammatik), wenn alle Produktionen die Form

(A, v) mit A ∈ V − T , v ∈ V ∗ − {ε}

besitzen.

1.2.2 Backus-Naur-Form und Syntaxdiagramm

Idee der Backus-Naur-Form:

Bei kontextfreien Sprachen sind die Regeln (A, v) mit A ∈ V − T

in der Form von Definitionsgleichungen A ::= v darstellbar.

Erweiterte Backus-Naur-Form (EBNF):

• Bildung syntaktischer Variabler:

Darstellung der Metasymbole durch Zeichenketten (Mnemotechnik),

z.B.

〈Ziffer〉, 〈Konstante〉, 〈Bedingte Anweisung〉, 〈Programm〉

• Definitionssymbol ::=

Die Regel (A, v) wird als Definitionsgleichung A ::= v

dargestellt, z.B.
〈Ziffer〉 ::= 3

〈Vorzeichen〉 ::= −
〈Vorzeichen〉 ::= +

〈Ganze Zahl〉 ::= 〈Vorzeichen〉〈Ziffernfolge〉

• Alternativsymbol |
Zusammenfassung mehrerer Regeln (A, v1), (A, v2), ..., (A, vn)

zu einer Definitionsgleichung

A ::= v1 | v2 | v3 | ... | vn |

z.B.
〈Vorzeichen〉 ::= + | −

〈Ziffer〉 ::= 0|1|2|3|4|5|6|7|8|9
〈Ziffernfolge〉 ::= 〈Ziffer〉 | 〈Ziffernfolge〉〈Ziffer〉



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 19

• Optionales Wiederholungssymbol { }
Eine durch { } geklammerte syntaktische Konstruktion kann be-

liebig oft stehen, muß jedoch nicht vorkommen, z.B.

〈Ziffernfolge〉 ::= 〈Ziffer〉 {〈Ziffer〉}

Beispiel 4 Sprache “Bezeichner“ G = (V, T, σ, P )

• T : A,B,C, ..., Y, Z, a, b, c, ..., z, 0, 1, 2, ..., 8, 9,

• V−T : 〈Ziffer〉, 〈Buchstabe〉, 〈Buchstabe oder Ziffer〉, 〈Bezeichner〉
• σ : 〈Bezeichner〉
• Syntax P :

〈Ziffer〉 ::= 0|1|2|3|4|5|6|7|8|9
〈Buchstabe〉 ::= A|B|C|...|Z|a|b|c|...|y|z

〈Buchstabe oder Ziffer〉 ::= 〈Buchstabe〉|〈Ziffer〉|
〈Bezeichner〉 ::= 〈Buchstabe〉{〈Buchstabe oder Ziffer〉}

Syntaxdiagramm (N.Wirth)

Das Syntaxdiagramm ist eine 2-dimensionale Darstellung der Syntax mit

folgenden Eigenschaften:

• Die Abarbeitung erfolgt von links nach rechts.

• Definierte Grundsymbole (terminals) erscheinen in ovalen Kästchen.

• Durch Syntaxdiagramme definierte syntaktische Variablen (non-terminals)

erscheinen in rechteckigen Kästchen.

• Alternativen | werden durch Verzweigungen realisiert.

• Wiederholungen { } werden durch Rückführungen realisiert.



Kapitel 2

Strukturierte Programmierung

Pascal ist eine allgemein anwendbare problemorientierte (imperative)

Programmiersprache, die ursprünglich von Niklaus Wirth entwickelt wur-

de und nach Blaise Pascal, dem berühmten Philosophen und Ma-

thematiker des 17. Jahrhunderts benannt wurde.

Historie:

• 1971 – erste Vorstellung der Sprache durch N. Wirth (TU Zürich)

• 1974 – Veröffentlichung des Sprachenreports (K. Jensen, N. Wirth)

• 1975 – Concurrent Pascal: für “Großrechner“ mit Multiprogramming

• 1983 – Turbo Pascal (TP): von BORLAND INT. für PC konzipiert

und mit effektiver Entwicklungsumgebung versehen

• 1986 – Object Pascal: kleine Erweiterung von Pascal um

wesentliche objektorientierte Sprachelemente

• 1992 – Borland Pascal (BP): Weiterentwicklung von Turbo Pascal

• 1991 – Pascal-XSC (PXSC): Pascal eXtensions for Scientific

Computation (U.Kulisch u.a., Uni Karlsruhe)

• 1990 – Standard: ISO Standard Pascal (ISO 7158)

• 1991 – Standard: ISO Extended Pascal Standard (ISO 10206)

• 1997 – GNU Pascal (GPC): GNU Project Pascal Compiler

(unter ständiger Weiterentwicklung)

• 2002 – Free Pascal (FP): Version 1.0.6 (unter Entwicklung)

Zielsetzungen:

• Systematisches Erlernen des strukturierten Programmierens

• Entwicklung von Implementationen, die zuverlässig und effektiv sind

• Universelle Sprache für viele Einsatzgebiete und Rechnertypen

2.1 Sprachelemente und Datentypen

2.1.1 Grundlegende Sprachelemente

Alphabet (Grundsymbole) von BP, FP, GPC :

〈Buchstabe〉 ::= A|B|C|...|Z|a|b|c|...|y|z
〈Ziffer〉 ::= 0|1|2|3|4|5|6|7|8|9

〈Sonderzeichen〉 ::= +| − | ∗ | = | < | > | . . . |〈Reserviertes Wort〉
〈Grundsymbol〉 ::= 〈Buchstabe〉|〈Ziffer〉|〈Sonderzeichen〉

Merke: Pascal unterscheidet nicht zwischen Groß- und Kleinbuchstaben,

es ist nicht “case sensitiv“!

〈Reserviertes Wort〉 ::= div|mod|or|and|not|xor|
if |then|else|for|do|while|
begin|end|const|var|type|...|
function|procedure|unit|program|

Merke: Möglich sind z.B. program, Program, PROGRAM

Begrenzer von BP, FP, GPC:

• Sie dienen der Trennung der Sprachelemente. Zwei Sprachelemente

müssen wenigstens von einem Begrenzer unterbrochen werden!

• Begrenzer sind (a) Leerzeichen, (b) Neue Zeile (CR), (c) Sonder-

zeichen außer reservierten Wörtern.

• Wo 1 Leerzeichen stehen kann, dürfen beliebig viele stehen!

2.1.2 Skalare Standardtypen

Ein Datentyp definiert die Art der Werte, die eine Variable annehmen

kann. Jede Variable in einem Programm muß genau einem Datentyp zu-

geordnet werden. Das geschieht durch eine Deklaration.

BP unterscheidet skalare Datentypen und strukturierte Datentypen. Letz-

tere werden aus den skalaren Typen aufgebaut.

Standard-Datentypen von BP, FP und GPC:

1. Natürliche Zahlen (unsigned-Typen):

byte, word, [cardinal, Qword, shortword, longword]

2. Ganze Zahlen (signed-Typen):

shortint, integer, [smallint, longint, int64, byteint]

3. Reelle Zahlen (float-Typen):

real, single, double, extended

4. Logische Wahrheitswerte (Boolean-Typen):

boolean mit Werten false, true (1 Byte)

5. Alphanumerische Zeichen (Character-Typen):

char mit Werten gemäß ASCII-Code (1 Byte)

Float-Standardtypen für reelle Zahlen (in BP, FP, GPC):

Typ Wertebereich Bytes Stellen

single 1.5 · 10−45 .. 3.4 · 1038 4 7 – 8
real 2.9 · 10−39 .. 1.7 · 1038 6 11 – 12
double 5.0 · 10−324 .. 1.7 · 10308 8 15 – 16
extended 1.9 · 10−4951 .. 1.1 · 104932 10 19 – 20

ANSI/IEEE - Standard 754 zur Gleitpunktarithmetik

Zahlenformate : single single double double
extended extended

Wortbreite 32 Bit ≥ 43 Bit 64 Bit ≥ 79 Bit
4 Byte ≥ 6 Byte 8 Byte ≥ 10 Byte

Mantissenbreite 24 Bit ≥ 32 Bit 53 Bit ≥64 Bit
Exponentenbreite 8 Bit ≥ 11 Bit 11 Bit ≥15 Bit
hidden bit ja undefiniert ja undefiniert
bias 127 undefiniert 1023 undefiniert

max. Exponent +127 ≥ +1023 +1023 ≥ +16383
min. Exponent −126 ≤ −1022 −1022 ≤ −16382

Dezimalstellen 6.9 − 7.2 ≥9.3 − 9.6 15.6 − 15.9 ≥18.9 − 19.2

größte Zahl 3.4 ∗ 10+38 ≥1.7 ∗ 10+308 1.7 ∗ 10+308 ≥1.1 ∗ 10+4932

kleinste norm. Zahl 1.2 ∗ 10−38 ≤2.3 ∗ 10−308 2.3 ∗ 10−308 ≤3.4 ∗ 10−4932

kl. denorm. Zahl 1.5 ∗ 10−45 ≤1.7 ∗ 10−317 5.0 ∗ 10−324 ≤3.7 ∗ 10−4951

Literatur :

1. ANSI: American National Standards Institute / IEEE: Institute of Electrical
& Electronic Engineers, New York. A Standard for Binary Floating-Point

Arithmetic, 1985. ANSI/IEEE Standard 754-1985.

2. ANSI: American National Standards Institute / IEEE: Institute of Electri-
cal & Electronic Engineers, New York. A Standard for Radix-Independent

Floating-Point Arithmetic, 1987. ANSI/IEEE Standard 854-1987.

Integer-Standardtypen (in BP, FP und GPC):

signed B Wertebereich Borland- GNU- Free-

Pascal Pascal Pascal

nein 1 0 .. 28 − 1 Byte Byte Byte

nein 2 0 .. 216 − 1 Word ShortWord Word

nein 4 0 .. 232 − 1 – Word Cardinal

nein 8 0 .. 264 − 1 – LongWord QWord

ja 1 −27 .. 27 − 1 ShortInt ByteInt ShortInt

ja 2 −215 .. 215 − 1 Integer ShortInt SmallInt

ja 4 −231 .. 231 − 1 LongInt Integer Integer

ja 8 −263 .. 263 − 1 Comp LongInt Int64

Bereichsgrenzen:
27 = 128

28 − 1 = 255

215 = 32768

216 − 1 = 65535

231 = 2 147 483 648

232 − 1 = 4 294 967 295

263 = 9 223 372 036 854 775 808

264 − 1 = 18 446 744 073 709 551 615

Äquivalente Bezeichnungen in GPC:

Byte = ByteCard

ShortWord = ShortCard

Word = Cardinal

LongWord = LongCard

2.2 Programmaufbau

Jedes Programm besteht aus dem Programmkopf, gefolgt vom Pro-

grammblock. Der (Programm-)Block setzt sich aus dem Deklarationsteil

und dem Anweisungsteil zusammen.

Beispiel 1 Kugelberechnung

Gegeben ist der Radius r = 5.12007 einer Kugel; gesucht sind Volu-

men und Oberfläche:

ofl = 4πr2 , vol =
4

3
πr3 =

1

3
ofl · r

Programm (1.Version):

Program Kugel;

{ Bestimmung von Oberflaeche und

Volumen einer Kugel mit Radius r }

var

r, ofl, vol : real;

begin

r := 5.12007;

ofl := 4.0 * pi * r * r;

vol := ofl / 3.0 * r

end.

Kommentare: Zur Dokumentation dürfen Kommentare überall im

Programm geschrieben werden, wo ein Begrenzer erlaubt ist. Sie sind in

Klammern { und } zu setzen!

2.2.1 Programmkopf

Syntax:

〈Programmkopf〉 ::= Program 〈Bezeichner〉
(〈Programmparameter〉);

〈Programmparameter〉 ::= 〈Bezeichner〉{, 〈Bezeichner〉}

Semantik:

• Der Programmname hinter Program sollte im Programmblock nicht

benutzt werden.

• Über Programmparameter kann das Programm mit seiner Umge-

bung kommunizieren. In ISO-Pascal sind sie obligatorisch; ggf. sind

die Standarddateien Input und Output zu benutzen.

• In BP können Programmparameter fehlen; dann werden Input und

Output gesetzt.

Beispiele:

Program Kugel;

Program Loesung_einer_Gleichung;

Program Sortieren (Input,Output);

Program Kopiere (Quellendatei, Zieldatei);

2.2.2 Deklarationsteil

Bedeutung der Deklaration:

Alle in einem Programm auftretenden Objekte – außer Standard-Objekten

– müssen zuerst deklariert (bekanntgemacht) werden, bevor sie im an-

schließenden Anweisungsteil verarbeitet werden können.

Objekt Deklarationsart Res. Wort

Variable Variablendeklaration Var

Konstante Konstantendeklaration Const

Datentyp Typdeklaration Type

Funktion Funktionsdeklaration Function

Verfahren Prozedurdeklaration Procedure

Sprungmarke Markendeklaration Label

Variablendeklaration:

Jede Variable muß vor ihrer Verwendung genau einmal deklariert werden.

Die Deklaration legt fest, zu welchem Datentyp die möglichen Werte der

deklarierten Variablen gehören. Die Deklaration dient der Bereitstellung

von Speicherplatz entsprechend dem angegebenen Typ.

Beispiele:

Var

i,j,index,k2 : integer;

a, radius_1, x,z : double;

p, q : Boolean;

zeichen : char;

Syntax (vorläufig):

〈Var-Deklaration〉 ::= 〈Bezeichner〉{, 〈Bezeichner〉} : 〈Typ〉
〈Var-Deklarationsteil〉 ::= Var 〈Var-Deklaration〉{; 〈Var-Deklaration〉};



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 28

Konstantendeklaration:

Im Konstantendeklarationsteil führt der Programmierer nach eigenem

Ermessen Namen (d.h. Synonyme) für Konstanten ein. Jeder auftreten-

de Konstantenbezeichner muß vor seiner Verwendung deklariert werden.

Ausnahmen sind Standardkonstanten (z.B. pi, maxint, true, false).

Man unterscheidet untypisierte Konstanten und typisierte Konstanten.

Beispiele (untypisierte Konstanten):

Const

n = 2048; { Mathemat. Konstanten }

e = 2.718281828459;

g = 9.81; { Physikal. Konstanten }

Boltzmann = 1.3804E-23;

Const { Weitere Konstanten }

PLZ = 98693;

Kennwort = ’SESAM’;

leer10 = ’ ’;

Syntax (vorläufig):

〈Const-Definition〉 ::= 〈Bezeichner〉 = 〈Konstante〉|
〈Bezeichner〉 : 〈Typ〉 = 〈Konstante〉

〈Const-Deklarationsteil〉 ::= Const 〈Const-Definition〉
{; 〈Const-Definition〉};

〈Konstante〉 ::= 〈vorzeichenlose Zahl〉|
〈Vorzeichen〉〈vorzeichenlose Zahl〉|
〈Konstantenbezeichner〉|
〈Vorzeichen〉〈Konstantenbezeichner〉|
〈Zeichenkette〉

BP läßt anstelle der Konstanten auch konstante Ausdrücke zu, die während

der Compilierung jedoch auswertbar sein müssen. Typisierte Konstanten

werden genau einmal beim Start des Programmes initialisiert.

Beispiele (typisierte Konstanten):

Const

n: Integer = 2048; { Mathemat. Konstante }

Boltz: Double = 1.3804E-23; { Physikal. Konstante }

Messg: String[20]= ’Hello girls’; { Zeichenkettentyp }

n2: Integer = n*n-1; { Konstanter Ausdruck }

Die Deklaration von Typen, Funktionen und Prozeduren wird später behandelt. Auf Marken wird
im Interesse des strukturierten Programmierens bewußt verzichtet.

2.2.3 Anweisungsteil

Der Anweisungsteil eines Programmes legt den Algorithmus fest, der

nach dem Programmstart abzuarbeiten ist. Er hat die Form einer Ver-

bundanweisung und folgt unmittelbar dem Deklarationsteil.

Syntax:

〈Anweisungsteil〉 ::= 〈Verbundanweisung〉
〈Verbundanweisung〉 ::= Begin 〈Anwweisung〉

{, 〈Anweisung〉}End.

Merke: Das Semikolon dient der Trennung von Anweisungen, nicht aber

deren Abschluß!

Beispiel (s.o.): begin

r := 5.12007;

ofl := 4.0 * pi * r * r;

vol := ofl / 3.0 * r

end.

Zuweisungs-Anweisung (Ergibtanweisung, Wertzuweisung)

Als grundlegendste Anweisung dient sie dazu, den berechneten Wert

eines Ausdruckes einer Variablen zuzuweisen. In Pascal fungiert := als

Zuweisungsoperator.

Beispiele:

var

i,k : integer;

r, ofl, vol : real;

x, y, z : double;

p, q : Boolean;

begin

{integer-Zuweisungen}

i := 1;

k := k+2;

{float-Zuweisungen}

ofl := 4.0 * pi * r * r;

vol := ofl / 3.0 * r;

z := x / y / z;

{Boolean-Zuweisung}

p := p and q or p and not q;

{Gemischte Zuweisungen}

x := i mod k;

z := x / r / y;

x := (4 + x) * (x*x - r) - 1;

end.

Merke: Typ der Variablen und Typ des Ausdruckes müssen übereinstim-

men. Ausnahmen bilden “miteinander verträgliche Typen“, z.B. wenn der

Typ des Ausdruckes ein Teilbereich des Variablentyps ist.

2.2.4 Ein- und Ausgabe, Geradeausprogramm

Ein- und Ausgabe über Standarddateien

• Eingabe über die Standarddatei Input mit den Prozeduren

read – Eingabe von Zeichen,

readln – Eingabe einer Zeile.

• Ausgabe über die Standarddatei Output mit den Prozeduren

write – Ausgabe von Zeichen,

writeln – Ausgabe einer Zeile.

Beispiel 1 Kugelberechnung (2.Version)

Program Kugel;

{ Bestimmung von Oberflaeche und

Volumen einer Kugel mit Radius r }

var

r, ofl, vol : real;

begin

writeln (’Kugelberechnung’);

writeln (’===============’);

writeln;

write (’Radius r = ’);

readln (r);

ofl := 4.0 * pi * r * r;

vol := ofl / 3.0 * r;

writeln (’Oberflaeche ofl = ’, ofl);

writeln (’Volumen vol = ’,vol);

readln

end.

• Eingabeanweisung read:

Die allgemeine Form lautet

read ( var_1, var_2, ... , var_n );

mit Variablenbezeichnern var_1, var_2, ... , var_n vom

Integer-Typ, Float-Typ, Char-Typ oder String-Typ. Die Anweisung

wird wie die Folge

read (var_1); read(var_2); ... ; read(var_n );

ausgeführt.

• Eingabeanweisung readln:

Die allgemeine Form

readln ( var_1, var_2, ... , var_n ); oder readln;

ist äquivalent der Folge

read (var_1); ... ; read(var_n ); readln;

Allgemeine Wirkung von readln : Übergang zu einer neuen Zeile

in der Eingabedatei. Bei Tastatureingabe wartet der Computer auf

die Eingabe eines Zeileabschlusses (End-of-Line, CR/LF), was durch

ENTER bewirkt wird.

Beispiele: readln (Name, Konto-Nr, Guthaben);

read (r, Dichte);

readln;

• Ausgabeanweisungen write und writeln:

Die allgemeine Form lautet

write ( par_1, par_2, ... , par_n ); bzw.

writeln ( par_1, par_2, ... , par_n );

mit Ausgabeparametern par_1, par_2, ... , par_n vom

Integer-Typ, Float-Typ, Char-Typ, Boolean-Typ oder String-Typ.

Die Anweisungen werden wie die Folgen

write (par_1); ... ; write(par_n ); bzw.

write (par_1); ... ; write(par_n ); writeln;

ausgeführt.

Allgemeine Wirkung von writeln : Schreiben eines Zeilenende-

Kennzeichens (EOLn, End-of-Line) in die Ausgabedatei. Bei Bild-

schirm- oder Druckerausgabe erfolgt ein Zeilenwechsel mit CR/LF.

• Ausgabeparameter: Die Syntax lautet:

〈Ausgabeparameter〉 ::= 〈Ausdruck1〉|
〈Ausdruck1〉 : 〈Ausdruck2〉|
〈Ausdruck1〉 : 〈Ausdruck2〉 : 〈Ausdruck3〉

Semantik:

– 〈Ausdruck1〉 stellt den auszugebenden Wert dar. Steht er ohne

weitere Parameter, so wird die Information in einer standardi-

sierten Form (ohne Informationsverlust) ausgegeben.

– Die Datenfeld-Längenparameter 〈Ausdruck2〉 und 〈Ausdruck3〉
dienen der Formatierung; sie müssen vom Integer-Typ sein.

– Format 〈Ausdruck2〉 gibt die gesamte Datenfeldlänge an. Der

Eintrag erfolgt rechtsbündig.

– 〈Ausdruck3〉 dient ausschließlich bei Float-Typen der (rechtsbündi-

gen) Festpunktdarstellung mit Rundung zur Angabe der gebro-

chenen Nachpunkt-Stellen.

Anweisung Ausgabe

write (1E0); 1.00000000000000E+0000
write (-pi); -3.14159265358979E+0000
write (-pi:20); -3.14159265359E+0000
write (-pi:17); -3.14159265E+0000
write (-pi:11); -3.14E+0000
write (-pi:10:6); -3.141593

Beispiel 2 Transformation von Kugelkoordinaten

Die Untersuchung kugelförmiger Bauteile wird oft durch sog. Kugelkoor-

dinaten erleichtert. Jeder Punkt P 6= 0 des 3D-Raumes wird eindeutig

durch folgende 3 Koordinaten beschrieben:

1. Radius r ≥ 0 für den Abstand zum Nullpunkt

2. Winkel ϕ zur positiven x-Achse mit −π < ϕ ≤ π

3. Winkel θ zur positiven z-Achse mit 0 ≤ θ ≤ π

Ein einzugebender Punkt P = (r, ϕ, θ) in Gradmaß ist in cartesische

Koordinaten P = (x, y, z) zu transformieren. Zusätzlich sollen die

Winkel θx und θy des Radiusvektors mit der positiven x- bzw. y-Achse

bestimmt werden.

Berechnungsformeln:

x = r cosϕ sin θ

y = r sinϕ sin θ

z = r cos θ

θx =
π

2
− arctan

x
√
r2 − x2

θy =
π

2
− arctan

y
√

r2 − y2

Program Kugelkoordinaten;

uses crt; { Einbinden der Bildschirmprozeduren }

var r, phi, theta, sig, x, y, z, thx, thy : double;

begin

clrscr; { Bildschirmloeschen }

writeln (’Winkel und (x,y,z)-Koordinaten’);

writeln (’==============================’);

write (’Radius (in cm) = ’); readln (r);

write (’PHI (in Grad) = ’); readln (phi);

write (’THETA (in Grad) = ’); readln (theta);

{ Umrechnung in Bogenmass }

phi := pi * phi / 180.0;

theta := pi * theta / 180.0;

{ Winkelberechnung }

sig := sin (theta);

x := sig * cos (phi);

y := sig * sin (phi);

thx := pi / 2.0 - arctan (x / sqrt (1.0 - sqr (x)));

thy := pi / 2.0 - arctan (y / sqrt (1.0 - sqr (y)));

writeln (’THETAX = ’, thx * 180.0 / pi:7:2, ’ Grad’);

writeln (’THETAY = ’, thy * 180.0 / pi:7:2, ’ Grad’);

{ Kartesische Koordinaten (x,y,z) }

x := r * x;

y := r * y;

z := r * cos (theta);

writeln (’Kartesische Koordinaten:’);

writeln (’x = ’:15, x);

writeln (’y = ’:15, y);

writeln (’z = ’:15, z); readln

end.

#include <iostream.h>

#include <math.h>

int main() {

double r, phi, theta, sig, x, y, z, thx, thy;

cout << "\nKugel- und (x,y,z)-Koordinaten" ;

cout << "\n==============================" << endl;

cout << "Radius (in cm) = "; cin >> r;

cout << "PHI (in Grad) = "; cin >> phi;

cout << "THETA (in Grad) = "; cin >> theta;

/* Umrechnung in Bogenmass */

phi = M_PI * phi / 180.0;

theta = M_PI * theta / 180.0;

/* Winkelberechnung */

sig = sin(theta);

x = sig * cos(phi);

y = sig * sin(phi);

thx = M_PI / 2.0 - atan(x / sqrt(1.0 - x*x));

thy = M_PI / 2.0 - atan(y / sqrt(1.0 - y*y));

cout << "THETAX = " << thx * 180.0 / M_PI << endl;

cout << "THETAY = " << thy * 180.0 / M_PI << endl;

/* Kartesische Koordinaten (x,y,z) */

x = r * x;

y = r * y;

z = r * cos(theta);

cout << "\nKartesische Koordinaten:" << endl;

cout << " x = " << x << endl;

cout << " y = " << y << endl;

cout << " z = " << z << endl;

return 0;

}

Standardfunktionen (Auswahl in BP, FP, GPC):

Aufruf Bedeutung Param.-typ Funkt.-typ

abs(x) |x| int, float int, float

sqr(x) x2 int, float int, float

sqrt(x)
√
x, x ≥ 0 int, float float

exp(x) ex int, float float

ln(x) lnx, x > 0 int, float float

sin(x) sinx, BM int, float float

cos(x) cosx, BM int, float float

arctan(x) arctanx, BM int, float float

int(x) ganzer Anteil float float

frac(x) gebrochener Anteil float float

round(x) Rundung auf ganz float integer

trunc(x) Abschneiden float integer

ord(x) ordinaler Wert Skalar-Typ integer

chr(x) Zeichen mit Ordinalwert x integer char

pred(x) Wert des Vorgängers Skalar-Typ Arg.-typ

succ(x) Wert des Nachfolgers Skalar-Typ Arg.-typ

odd(x) Ist x ungerade ? int Boolean

random(x) gleichverteilte Zufallszahl mit int int

random 0 ≤ z < x bzw. 0 ≤ z < 1 – real

upcase(x) Großgeschriebenes Äquivalent char char

keypressed Taste gedrückt ? – Boolean

sizeof(x) Speicherplatz für x (Bytes) beliebig integer

Abkürzungen: BM - Bogenmaß, int - integer-Typ, float - float-Typ

2.3 Kontrollstrukturen

Pascal besitzt folgende strukturierte Anweisungen:

• Verbundanweisung

• Bedingte Anweisungen (if-Anweisung, case-Anweisung)

• Zyklusanweisungen (while-, repeat-, for-Anweisung)

2.3.1 Verbundanweisung

Begin und end fungieren als Anweisungsklammern. Sie fassen eine

Sequenz zu einer einzigen Anweisung zusammen. Die Verbundanweisung

wird dort gebraucht, wo mehrere Anweisungen ausgeführt werden sollen,

die Syntax jedoch nur eine Anweisung gestattet.

Beispiel:

Zyklische Vertau-

schung dreier Größen

x, y, z

begin

hilf := z;

z := y;

y := x;

x := hilf

end;

2.3.2 Bedingte Anweisungen

If–Anweisung: Es existieren 2 Formen

• Verkürzte if-Anweisung – Ausführung einer Anweisung, falls eine

Bedingung erfüllt ist

• Vollständige if-Anweisung – Ausführung der Anweisung 1, falls eine

Bedingung erfüllt ist, andernfalls Ausführung der Anweisung 2

Syntax:

〈if -Anweisung〉 ::= if 〈Bedingung〉 then 〈Anweisung〉|
if 〈Bedingung〉 then 〈Anweisung 1〉
else 〈Anweisung 2〉

Beispiel 1 Quadratische Gleichung x2 + ax+ b = 0 .

Program Quadratische_Gleichung;

var

a, b, c, D, w, x1, x2, re, im : real;

begin

readln (a, b); c := -a / 2.0;

D := c * c - b; w := sqrt (abs (D));

if D < 0 then

begin

re := c; im := w;

writeln(re, im, ’Loesungen komplex’)

end

else

begin

x1 := c + w; x2 := c - w;

writeln (x1, x2, ’Loesungen reell’)

end

end.

Geschachtelte if–Anweisung:

Im then-Zweig oder im else-Zweig einer if-Anweisungen treten häufig

weitere if-Anweisungen auf.

Merke: Eine else-Klausel gehört stets zu dem nach links nächsten (in-

nersten) if, das noch keine else-Klausel hat (“else-Regel“).

Beispiel:

if a > b then if a > c then M := a else M := c;

Richtige Interpretation:

if a > b then

if a > c then M := a

else M := c;

Falsche Interpretation:

if a > b then

if a > c then M := a

else

M := c;

Beispiel 2 Maximum max(a, b, c) dreier Zahlen

Program Maxi;

var

a, b, c, max : double;

begin

readln(a); readln(b); readln(c);

if (a > b) and (a>c) then max := a

else

if c > b then max := c

else max := b ;

writeln (’MAX = ’, max); readln

end.

Case–Anweisung:

Sie legt fest, daß aus endlich vielen Anweisungen genau eine auszuführen

ist. In Abhängigkeit vom Wert eines Selektors wird entschieden,

– ob eine der aufgelisteten Anweisungen bzw.

– welche dieser Anweisungen

ausgeführt wird.

Struktur der case-Anweisung:

case 〈Selektor〉 of
〈Auswahlmarken 1〉 : 〈Anweisung 1〉;
〈Auswahlmarken 2〉 : 〈Anweisung 2〉;
. . . . . .

〈Auswahlmarken n〉 : 〈Anweisung n〉
else

〈Anweisung n+1〉;
〈Anweisung n+2〉;
. . . . . .

〈Anweisung n+m〉
end

Beispiel 3 Auswahl aus 4 Funktionen

In Abhängigkeit von nr ∈ {1, 2, 3, 4} sind die Funktionswerte

f1(x) = sinx

f2(x) = cosx

f3(x) = sinhx = (ex − e−x)/2

f4(x) = coshx = (ex + e−x)/2

zu berechnen, wenn x eingegeben wird.

Beispiel 3 Auswahl aus 4 Funktionen

In Abhängigkeit von nr ∈ {1, 2, 3, 4} sind die Funktionswerte

f1(x) = sinx

f2(x) = cosx

f3(x) = sinhx = (ex − e−x)/2

f4(x) = coshx = (ex + e−x)/2

zu berechnen, wenn x eingegeben wird.

Program Funktionen;

var

nr : integer;

x,y : double;

begin

readln(nr); readln(x);

case nr of

1 : y := sin (x);

2 : y := cos (x);

3 : begin

y := exp (x);

y := (y - 1.0 / y) / 2.0

end;

4 : begin

y := exp (x);

y := (y + 1.0 / y) / 2.0

end

end;

writeln(x,y); readln

end.



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 43

// Version des Beispiels 3 in C++

#include <iostream.h>

#include <math.h>

int main() {

long nr;

double x, y = 1.0E20;

/* Eingabe nr und x */

cout << "\nFunktionsauswahl" ;

cout << "\n================" << endl;

cout << "Funktions - Nr = "; cin >> nr;

cout << "Argument x = "; cin >> x;

switch (nr)

{

case 1:

y = sin(x); break;

case 2:

y = cos(x); break;

case 3:

y = exp(x);

y = (y - 1.0 / y) / 2.0; break;

case 4:

y = exp(x);

y = (y + 1.0 / y) / 2.0; break;

}

cout << "Funktionswert y = " << y << endl;

return 0;

}

2.3.3 Zyklusanweisungen

While–Anweisung:

Die while-Anweisung wirkt abweisend, d.h. zuerst wird eine Bedingung

ausgewertet. Ist deren Wert true , so wird der Schleifenkörper (An-

weisung) ausgeführt und anschließend erneut die Bedingung getestet. Im

anderen Falle wird sofort zur nächsten Programmanweisung übergegan-

gen.

Syntax:
〈while-Anweisung〉 ::= while 〈Bedingung〉

do 〈Anweisung〉

Beispiele:

while x > 0.1 do x := 0.5*x;

while not q do

begin

readln(x);

q := abs(x-1.0) < epsilon

end;

PAP: Struktogramm:

Beispiel 4 Bisektionsverfahren für f(x) = 0

Gegeben sind eine Funktion f(x) = sinx − e−x und ein Inter-

vall [a, b] mit a < b und f(a) · f(b) < 0 . Gesucht ist eine

Nullstelle x ∈ [a, b] mit einer relativen Genauigkeit (Toleranz) ε > 0.

Program Bisektion;

{ Funktion f(x) = sin(x) - exp(-x) }

var

a, b, epsi, x, y : double;

begin

writeln (’Bisektionsverfahren’);

readln (a); readln (b); readln (epsi);

y := sin (a) - exp (-a);

if y >= 0.0 then

begin

x := a; a:= b; b := x

end; { Nun ist f(a)<0 und f(b)>0 }

while abs (a - b) > epsi * abs (a) do

begin

x := (a +b) / 2.0;

y := sin (x) - exp (-x);

if y < 0.0 then a := x else b := x

end; { while-Zyklus }

writeln (’Nullstelle x = ’,x); readln

end.



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 46

// Version des Beispiels 4 in C++

#include <iostream.h>

#include <math.h>

int main() {

/* Funktion f(x) = sin(x) - exp(-x) */

double a, b, epsi, x, y;

cout << "\nBisektionsverfahren" << endl;

cout << "===================" << endl;

cout << "Linkes Intervallende a = "; cin >> a;

cout << "Rechtes Intervallende b = "; cin >> b;

cout << "Genauigkeitsschranke epsi = "; cin >> epsi;

y = sin(a) - exp(-a);

if (y >= 0.0)

{

x = a; a = b; b = x;

} // Nun ist f(a)<0 und f(b)>0

while (fabs(a - b) > epsi * (fabs(a)+ 1.0))

{

x = (a + b) * 0.5;

cout << x << endl;

y = sin(x) - exp(-x);

if (y < 0.0)

a = x;

else

b = x;

} /* while - Zyklus */

cout << "\nNullstelle x = " << x << endl;

return 0;

}

Repeat–Anweisung:

Die repeat-Anweisung ist nicht abweisend, d.h. zuerst wird der Schlei-

fenkörper (Anweisungs-Sequenz) mindestens einmal ausgeführt und an-

schließend die Bedingung getestet. Ist sie erfüllt, so wird die repeat-

Anweisung beendet, andernfalls wird der Schleifenkörper erneut abgear-

beitet.

Syntax:

〈repeat-Anweisung〉 ::= repeat 〈Anweisung〉{; 〈Anweisung〉}
until 〈Bedingung〉

Beispiele:

repeat x := 0.5*x

until x < 1.0;

repeat

readln(x);

q := abs(x-1.0) < epsilon

until q;

PAP: Struktogramm:

Beispiel 5 Quadratwurzelbestimmung
√
x , x > 0

Mit dem Newton–Verfahren kann man y =
√
x durch die quadratisch

konvergente Iterationsvorschrift

yn+1 =
1

2
(yn +

x

yn
) , n = 0, 1, 2, ...

bis auf eine relative Genauigkeit ε bestimmen. Nach jedem Schritt wird

geprüft, ob

|yn+1 − yn| < ε|yn+1|
gilt.

Program Quadratwurzel;

{ sqrt (x) fuer x > 0 }

var

x, eps, y, ya : double;

begin

write (’x = ’); readln (x);

write (’eps = ’); readln (eps);

y := x;

repeat

ya := y;

y := (ya + x / ya) / 2.0

until abs (y - ya) < eps * abs (y);

writeln (’y = ’, y); readln

end.

For–Anweisung:

Die for-Anweisung (Laufanweisung) nutzt man für Zyklen, die durch ei-

ne Laufvariable (Kontrollvariable) lv gesteuert werden. Diese Variable

durchläuft einen Laufbereich von einem Anfangswert aw bis zu einem

Endwert ew.

Syntax:

〈for-Anweisung〉 ::= for 〈Laufvariable〉 := 〈Laufliste〉
do 〈Anweisung〉

〈Laufliste〉 ::= 〈Anfangswert〉 to 〈Endwert〉|
〈Anfangswert〉 downto 〈Endwert〉

Beispiele:

for n := -3 to 3 do x := x+2;

for n := 3 downto -1 do y := y+1;

for k := n-1 to 2*n+1 do z := k*z;

for c := ’a’ to ’z’ do write(c);

Merke:

• lv, aw und ew müssen vom selben Typ sein; dieser einfache Typ darf

kein float-Typ sein.

• Vor Abarbeitung der for-Anweisung werden die Werte von aw und

ew berechnet.

• Der Wert von lv darf nicht durch die Anweisung zusätzlich verändert

werden.

• Die for-Anweisung ist abweisend, d.h. zuerst wird die Zulässigkeit

der Laufvariablen geprüft und anschließend ggf. der Schleifenkörper

ausgeführt.

Beispiel 6 Arithmetisches Mittel von Meßwerten

Gegeben sind n reelle Meßwerte a1, a2, . . . , an einer technischen Größe

a . Beim Einlesen ist zu prüfen, ob ai > 50 ist; derartige “Ausreißer“

sind bei der Mittelbildung nicht zu berücksichtigen:

s :=
1

m

n
∑

i=1

|ai|≤50

ai , m > 0

Program Mittel;

var

i, n, m : integer;

s, a : double;

begin

write (’Anzahl n = ’); readln (n); writeln;

m := 0; s := 0.0;

for i := 1 to n do

begin

write (’a(’,i,’) = ’); readln (a);

if abs (a) <= 50.0 then

begin

s := s + a;

m := m + 1

end

end;

writeln;

if m = 0 then

writeln (’Das Mittel s ist nicht bestimmbar!’)

else begin

s := s / m; writeln (’Mittel s = ’,s)

end; readln

end.

Geschachtelte Zyklusanweisungen:

Eine Zyklusanweisung kann komplett im Schleifenkörper einer ande-

ren Zyklusanweisung liegen (geschachtelte Zyklen). Die Schleifenkörper

dürfen dann einander nicht überschneiden. Geschachtelte for-Anweisungen

dürfen nicht ein- und dieselbe Laufvariable besitzen.

Beispiel 7 Tabelle der Binomialkoeffizienten

Binomialkoeffizienten werden insbesondere bei statistischen Untersuchun-

gen benötigt. Zu einer natürlichen Zahl nmax sind alle Binomialkoeffi-

zienten
(

n

k

)

=
n

k!(n− k)!
=

n

1
· n− 1

2
· n− 2

3
· · · n− k + 1

k
·

für k = 0(1)n, n = 0(1)nmax zu berechnen und in einer Tabelle

auszugeben:

n = 0 k = 0 bk = 1

n = 1 k = 0 bk = 1

k = 1 bk = 1

n = 2 k = 0 bk = 1

k = 1 bk = 2

k = 2 bk = 1

n = 3 k = 0 bk = 1

k = 1 bk = 3

k = 2 bk = 3

k = 3 bk = 1

usw.

Program Binomialkoeffizienten;

var

nmax, n, k, i, kk : integer;

bk : double;

begin

write (’nmax = ’); readln (nmax); writeln;

for n := 0 to nmax do

begin

for k := 0 to n do

begin

if k > n - k then kk := n - k

else kk := k;

bk := 1.0;

for i := 1 to kk do

bk := bk * (n - i + 1) / i;

if k = 0 then write (’n=’:3, n:3)

else write (’ ’:6);

writeln (’k=’:8,k:3,’bk=’:9,bk:20)

end; { k - Zyklus }

writeln

end { n - Zyklus }

end.

/* Output from p2c, the Pascal-to-C translator */

/* From input file "binkoef2.pas" */

#include <iostream.h>

int main()

{

long nmax, n, k, i, kk;

double bk;

cout << endl << "nmax = "; cin >> nmax;

for (n = 0; n <= nmax; n++)

{

for (k = 0; k <= n; k++)

{

if (k > n - k)

kk = n - k;

else

kk = k;

bk = 1.0;

for (i = 1; i <= kk; i++)

bk = bk * (n - i + 1) / i;

if (k == 0)

cout << "n = " << n;

else

cout << " ";

cout << " k = " << k << " bk = " << bk << endl;

} // k - Zyklus

cout << endl;

} // n - Zyklus

return 0;

}

2.4 Prinzipien der Programmentwicklung

Bisher wurden einfache Probleme betrachtet, deren Algorithmierung,

Programmierung, Implementierung und Erprobung durch 1 Person aus-

geführt werden konnte.

Bei umfangreichen Projekten der Ingenieurpraxis ist jedoch eine Korre-

spondenz zwischen Auftraggeber, Programmentwickler und Programm-

nutzer erforderlich, um

– die Struktur der Ein- und Ausgabedaten

– die Struktur des Programmes und seiner Teile

endgültig festzulegen.

Forderung = Systematisierung des Entwicklungsprozesses:

(1) Anforderungsdefinition: Festlegung der Aufgabenstellung

(2) Problemanalyse: Einarbeitung in das Problem, Algorithmierung we-

sentlicher Teilschritte

(3) Festlegung der logischen Programmstruktur: Struktogramme, Pro-

grammablaufplan o.a. Schemata

(4) Entwicklung der physischen Programmstruktur: Programm, Module

und Unterprogramme

(5) Programmerprobung: Verifikation, Optimierung

(6) Programmanalyse: Rechnung von Alternativen und Aufwandsver-

gleiche

Es empfiehlt sich - auch bei kleineren Problemen - bereits diese Schritte

zu berücksichtigen.

Beispiel 7 Bessel-Funktion J0(x)

Die Schwingungen einer kreisförmigen, am Rand eingespannten Mem-

bran mit Radius a werden durch eine partielle Differentialgleichung be-

schrieben, deren Lösung mit Hilfe der Bessel-Funktionen 1.Art Jk(x),

k = 0, 1, 2, ... notiert werden können. Diese höheren transzendenten

Funktionen lassen sich durch Potenzreihen definieren:

Jk(x) =
∞
∑

n=0

(−1)n · x2n+k

22n+k · n! · (k + n)!
, k = 0, 1, 2, 3, ...

Aufgabe: Man erzeuge zu einzugebender Schrittweite h > 0 eine Ta-

belle der Bessel-Funktion J0(x) für x = 0(h)6.0 .

Problemanalyse:

• Darstellung von J0(x) :

J0(x) = 1 − 1

22 · 1
x2 +

1

24 · 22
x4 − 1

26 · 62
x6 + − · · ·

• Frage: Wieviele Glieder sind zu berücksichtigen?

mit an :=
(−1)n · x2n

22n · (n!)2
gilt lim

n→∞
an = 0

Abbruch der Summation, falls |an| < 10−10 ist.

• Effektive Berechnung der Summanden an :

a0 = 1 , an+1 = −
[

x

2(n+ 1)

]2

· an , n = 0, 1, 2, ...

Logische Programmstruktur: Struktogramm

Physische Programmstruktur: 1. Version des Programmes

{ Programm Besselfunktion_J0 }

{ Deklarationen }

{ Eingabe h, Ausgabe Tabellenkopf }

x := 0; xende := 6.0 + 0.1 * h;

while x <= xende do begin

{ Berechnung von y = J0(x) }

{ Ausgabe von x,y }

x := x + h

end { while x }

{ Programmende Besselfunktion_J0 }

Physische Programmstruktur: 2. Version des Programmes

Program Besselfunktion_J0;

{ Deklarationen }

{ Eingabe h, Ausgabe Tabellenkopf }

x := 0; xende := 6.0 + 0.1 * h;

while x <= xende do begin

{ Berechnung von y = J0(x) }

a:=1.0; n := 0; s := 1.0;

while abs (a) >= 1e-10 do begin

n := n+1;

a := - sqr (x / 2.0 / n) * a;

s := s + a

end; { while a }

y := s;

{ Ausgabe von x,y }

x := x + h

end { while x }

end. { Programmende Besselfunktion_J0 }

Physische Programmstruktur: Endversion des Programmes

Program Besselfunktion_J0;

var { Deklarationen }

h, x, xende, a, s, y :real;

n : integer;

begin

{ Eingabe h, Ausgabe Tabellenkopf }

writeln (’Tabelle der Besselfunktion J0(x)’);

writeln (’================================’);

write(’Schrittweite h = ’); readln(h); writeln;

writeln (’x’:10, ’J0(x)’:8);

x := 0; xende := 6.0 + 0.1 * h;

while x <= xende do

begin

{ Berechnung von y = J0(x) }

a:=1.0; n := 0; s := 1.0;

while abs (a) >= 1e-10 do

begin

n := n+1;

a := - sqr (x / 2.0 / n) * a;

s := s + a

end; { while a }

y := s;

{ Ausgabe von x,y }

writeln (x, y); x := x + h

end { while x }

end. { Programmende Besselfunktion_J0 }

Hinweis: Für komplexere Projekte hat sich das Entwurfsprinzip der

schrittweisen Verfeinerung (Top–down–Entwurf) bewährt.

Endversion des Programmes in C++

#include <iostream.h>

#include <math.h>

int main() {

double h, x, xende, a, s, y;

long n;

/* Eingabe h, Ausgabe Tabellenkopf */

cout << "Tabelle der Besselfunktion J0(x)" << endl;

cout << "================================" << endl;

cout << "Schrittweite h = "; cin >> h;

cout << endl << "x J0(x)" << endl;

x = 0.0; xende = 6.0 + 0.1 * h;

while (x <= xende)

{

/* Berechnung von y = J0(x) */

a = 1.0; n = 0; s = 1.0;

while (abs(a) >= 1e-10)

{

n++;

a = - (x / 2.0 / n) * (x / 2.0 / n) * a;

s += a;

}

y = s;

cout << x << " " << y << endl;

x += h;

}

return 0;

}



Kapitel 3

Prozedurale Programmierung

Um Programme zu entwickeln, die den Prinzipien der Universalität (Nut-

zung allgemeiner Grundprinzipien, leichte Anwendbarkeit) und der Un-

abhängigkeit (Lauffähigkeit auf verschiedenen Computern, Einsetzbar-

keit in andere Programme) genügen, wurde das Konzept der Prozedur

entwickelt.

Definition 1 : Prozedur 1

Eine Prozedur ist eine Folge von Anweisungen (Aktionen), die einen

in sich abgeschlossenen Programmteil darstellt, der von einem anderen

Programmteil aus aufgerufen, d.h. ausgeführt werden kann. Eine Pro-

zedur hat eine Schnittstelle, die ihren Namen, die zulässigen Eingabe-

objekte (auch Argumente genannt) und das Resultat, d.h. die nach der

Ausführung vorliegenden Resultatobjekte, spezifiziert.

Das Konstrukt “Prozedur“ ist in der prozedurorientierten (proze-

duralen) Programmierung von zentraler Bedeutung.

Pascal unterscheidet folgende 2 Arten von Unterprogrammen (Routinen):

• Verfahrens-Unterprogramme ⇒ Prozeduren

• Funktions-Unterprogramme ⇒ Funktionen

1Rechenberg,P.; Pomberger,G.(Hrsg.): Informatik-Handbuch. 2. Auflage. Hanser Verlag 1999, S.518

3.1 Prozeduren in PASCAL

3.1.1 Prozedur–Deklaration

Die Prozedur-Deklaration legt (a) die Eingangs- und Ausgangsparame-

ter und (b) den Inhalt des Algorithmus fest. Damit dient sie zugleich der

Prozedur-Definition. Jede Prozedur besteht - wie ein Programm - aus

dem Prozedurkopf, gefolgt vom Prozedurblock. Der (Prozedur-)Block

setzt sich aus dem Deklarationsteil und dem Anweisungsteil zusammen.

(a) Prozedurkopf

Syntax (vorläufig):

〈Prozedurkopf〉 ::= Procedure 〈Prozedurname〉
(〈Prozedurparameter〉);

〈Prozedurparameter〉 ::= 〈Par-Abschnitt〉{; 〈Par-Abschnitt〉}
〈Par-Abschnitt〉 ::= Var 〈Variable〉{, 〈Variable〉} : 〈einfacher Typ〉

Semantik:

• Über Prozedurparameter kann die Prozedur mit ihrer Umgebung

kommunizieren. Man unterscheidet Eingangs- und Ausgangspara-

meter.

• Der Prozedurname hinter Procedure darf im nachfolgenden Pro-

zedurblock benutzt werden ⇒ rekursive Prozedur.

• In BP können Prozedurparameter fehlen ⇒ parameterlose Prozedur.

Beispiele:

Procedure zeile (var n : integer; var ch : char);

{ n, ch - Eingangsparameter; keine Ausgangsparameter }

Procedure Bisektions_Verfahren (var a,b,epsi,x : double);

{ a,b,epsi - Eingangsparameter; x - Ausgangsparameter }

Procedure tausche (var a,b : single);

{ a,b - Eingangsparameter; a,b - Ausgangsparameter }

(b) Prozedurblock

Er besteht - wie ein Programmblock - aus Deklarationsteil und Anwei-

sungsteil. Falls nichts zu deklarieren ist, kann der Deklarationsteil fehlen.

Merke: Auch für Prozeduren gilt, daß alle dort auftretenden Objekte –

außer Standard-Objekten und Prozedurparametern – zuerst deklariert

(bekanntgemacht) werden müssen, bevor sie im anschließenden Anwei-

sungsteil verarbeitet werden können!

Beispiel 1 Ausgabe von n Zeichen

Es sind n identische Zeichen der Art ch hintereinander in einer Zeile

auszugeben, z.B. ’**********’ oder ’ ’ (5 Leerzeichen).

Procedure zeile (var n : integer; var ch : char);

{ n, ch - Eingangsparameter }

var

i : integer; { lokale "Hilfsvariable" }

begin

for i:=1 to n do

write(ch)

end;

Beispiel 2 (s.o.) Bisektionsverfahren für f(x) = 0

Gegeben sind eine Funktion f(x) = sinx − e−x und ein Inter-

vall [a, b] mit a < b und f(a) · f(b) < 0 . Gesucht ist eine

Nullstelle x ∈ [a, b] mit einer relativen Genauigkeit (Toleranz) ε > 0.

Procedure Bisektions_Verfahren

(var a, b, epsi, x : double);

{ a,b,epsi - Eingangsparameter;

x - Ausgangsparameter }

var

y : double; { lokaler Parameter }

begin

y := sin (a) - exp (-a);

if y >= 0.0 then

begin

x := a; a:= b; b := x

end; { Nun ist f(a)<0 und f(b)>0 }

while abs (a - b) > epsi * abs (a) do

begin

x := (a +b) / 2.0;

y := sin (x) - exp (-x);

if y < 0.0 then a := x else b := x

end { Nullstelle ist x }

end;

Auftretende Parameterarten:

• Prozedurparameter = formale Parameter (n,ch bzw. a,b,epsi,x):

Sie dienen der Übergabe von Eingangs- und Ausgangswerten zum

aufrufenden Programm. Da die konkreten (aktuellen) Werte zum

Zeitpunkt der Deklaration noch nicht bekannt sind, dienen sie als

“Platzhalter“ (deshalb: formale Parameter).

• Hilfsparameter = lokale Parameter (i bzw. y):

Sie werden im Deklarationsteil der Prozedur vereinbart und sind des-

halb auch nur im Innern der Prozedur gültig und verfügbar (deshalb:

lokale Parameter).

3.1.2 Prozedur–Aufruf

Prozeduren werden im Deklarationsteil eines Programmes (oder einer

Prozedur) vereinbart. Bei BP kann dies an beliebiger Stelle des Dekla-

rationsteil geschehen. Der Prozedur-Aufruf im Anweisungsteil aktiviert

den Prozedurblock und überträgt ihm die Steuerung des Programmab-

laufes. Dabei werden die formalen Parameter durch aktuelle Parameter

(Argumente) ersetzt.

Beispiele für Prozedur-Aufrufe:

zeile (m, zeichen);

zeile (n, ch);

Bisektions_Verfahren (a, b, Toleranz, Loesung);

Bisektions_Verfahren (u_Grenze, o_Grenze, epsilon, x0);

Merke:

• Der Aufruf enthält den Prozedurnamen und - in Klammern - die

Liste der aktuellen Parameter (der Argumente). Vorläufige Form

der aktuellen Parameter = Variablen.

• Beim Aufruf wird jeder aktuelle Parameter dem entsprechenden For-

malparameter der Reihenfolge nach zugeordnet.

• Folge: Anzahl und Typen der einander zugeordneten aktuellen und

formalen Parameter müssen übereinstimmen. Die entsprechenden

Bezeichner hingegen können sich voneinander unterscheiden.



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 64

Beispiel 1 (s.o.) Ausgabe von n Zeichen

Es sind n identische Zeichen der Art ch hintereinander in einer Zeile

auszugeben, z.B. ’**********’ bzw. ’ ’ (Leerzeichen).

Program Zeichen_Ausgabe;

var

m, n : integer;

ch, zeichen : char;

Procedure zeile (var n: integer;var ch: char);

{ n, ch - Eingangsparameter }

var

i : integer; { lokaler Parameter }

begin

for i:=1 to n do write(ch)

end; { of Procedure zeile }

begin

write(’Anzahl n = ’); readln(n);

ch := ’*’;

zeile (n,ch); writeln;

m := 54; zeichen := ’ ’; {Leerzeichen }

zeile (n,ch);

zeile (m,zeichen);

zeile (n,ch); readln

end.

Beispiel 2 (s.o.) Bisektionsverfahren für f(x) = 0

Einzugeben ist ein Intervall [a, b] mit a < b und eine Genauigkeits-

schranke (Toleranz) tol > 0. Die Nullstelle x ∈ [a, b] ist auszugeben.

Program Bisektion;

{ Funktion f(x) = sin(x) - exp(-x) }

var

e, f, tol, loesung : double;

Procedure Bisektions_Verfahren

(var a,b,epsi,x : double);

var

y : double; { lokaler Parameter }

begin

{ Prozedur-Anweisungen siehe oben

. . . . . . . . . . . . . . . . }

end;

begin { Hauptprogramm }

writeln (’Bisektionsverfahren’);

write(’Intervallgrenze e = ’); readln(e);

write(’Intervallgrenze f = ’); readln(f);

write(’Toleranz tol = ’); readln(tol);

Bisektions_Verfahren (e, f, tol, loesung);

writeln (’Nullstelle = ’,loesung); readln

end.

Bisektionsverfahren

===================

Intervallgrenze e = 0

Intervallgrenze f = 2

Toleranz tol = 1e-5

Nullstelle = 5.885 353 088 379E-1

Intervallgrenze e = 0

Intervallgrenze f = 2

Toleranz tol = 1e-10

Nullstelle = 5.885 327 440 337E-1

Intervallgrenze e = 0

Intervallgrenze f = 2

Toleranz tol = 1e-15

Nullstelle = 5.885 327 439 819E-1

3.1.3 Parameterübergabe

Parametervermittlung an obigem Beispiel:

Procedure

Bisektions_Verfahren(var a, b, epsi, x :double);

| | | |

1| 2| 3| 4|

| | | |

Bisektions_Verfahren( e, f, tol, loesung);

Eine Variable kann benutzt werden, wenn sie (a) einen Namen, (b) eine

Adresse und (c) einen Wert besitzt. Formale Parameter besitzen anfangs

jedoch nur Namen und Adresse (z.B. a, b, epsi).

2 Mechanismen der Wertvermittlung in PASCAL:

– Variablenaufruf (Namensaufruf, Referenzaufruf)

– Wertaufruf

(a) Variablenaufruf (Referenzaufruf, call by name)

• Kennzeichen: Reserviertes Wort var vor dem Formalparameter a

• Beispiele (s.o.):

Procedure Bisektions_Verfahren

(var a, b, epsi, x : double);

Procedure zeile (var n: integer;var ch: char);

• Realisierung: Übertragung der Adresse des aktuellen Parameters e

in den Werteteil des formalen Parameters a ⇒ Zugriff zum Wert

von a stets (indirekt) über die Adresse, d.h. über eine Referenz

• Bezeichnung: Variablenparameter, Referenzparameter

• Konsequenzen:

– Aktueller Parameter e muß eine Variable sein

– Jede Veränderung des formalen Parameters a verändert auch

den aktuellen Parameter e

– Anwendung bei Ergebnisparametern obligatorisch

– Nutzung um Speicherplatz zu sparen (große Vektoren und Ma-

trizen)

• Schematische Darstellung:

(b) Wertaufruf (call by value)

• Kennzeichen: Das reservierte Wort var fehlt vor dem Formalpara-

meter a

• Beispiele (s.o.):

Procedure Bisektions_Verfahren

( a, b, epsi : double; var x : double);

Procedure zeile ( n: integer; ch: char);

• Realisierung:

– Automatische Bildung einer lokalen Hilfsvariablen (no name)

– Berechnung des Wertes des aktuellen Parameters e und Zuwei-

sung zur Hilfsvariablen

– Direkte Verwendung der Hilfsvariablen (entspricht a) und ihres

Wertes im Prozedurblock

• Bezeichnung: Wertparameter (hier: a, b, epsi, n, ch)

• Konsequenzen:

– Aktueller Parameter e kann eine Variable, Konstante oder ein

Ausdruck sein

– Variablen des aufrufenden Programmes behalten ihren Wert; es

findet keine wertmäßige Veränderung statt

– Anwendung nur bei Eingangsparametern sinnvoll

Beispiele für Wertaufrufe:

zeile (25, ’*’);

readln(n); ch := ’ ’; zeile(2*n+1, ch);

readln(e); readln(f);

Bisektions_Verfahren (e+3.0*f, e+exp(f), 1E-8, Loesung);

Beispiel 3 Zyklische Vertauschung

Die Inhalte der 3 Variablen x, y, z sind zyklisch zu vertauschen:

x → y → z → x .

Program Zyklisches_Vertauschen;

var

x, y, z : double;

Procedure Tausche (var a, b, c : double);

{ a,b,c - Eingangs- und Ausgangsparameter }

var

hilf : double; { lokaler Parameter }

begin

hilf := c;

c := b;

b := a;

a := hilf

end;

begin

readln(x); readln(y); readln(z);

Tausche (x,y,z); { Variablenaufruf }

writeln (x,y,z); readln

end.

Frage: Was bewirkt die Prozedur Tausche, falls a,b,c Wertparameter

sind?

3.2 Funktionen in PASCAL

Eine Funktion ist in PASCAL eine Vorschrift zur Berechnung eines skalaren

Wertes aus einer oder mehreren Eingangsgrößen. Der Funktionstyp muß

also skalar sein!

3.2.1 Funktions–Deklaration

Die Funktions-Deklaration legt (a) die Eingangsparameter und deren

Typ, (b) den Ergebnistyp und (c) den Inhalt des Algorithmus fest.

Aufbau der Funktions-Deklaration

Sie ist wie die Prozedur-Deklaration aufgebaut - mit folgenden 2 Abwei-

chungen:

• Syntax des Funktionskopfes:

〈Funktionskopf〉 ::= function 〈Funktionsname〉
(〈Funktionsparameter〉) : 〈einfacher Typ〉;

〈Funktionsparameter〉 ::= 〈Par-Abschnitt〉{; 〈Par-Abschnitt〉}
〈Par-Abschnitt〉 ::= Var 〈Variablenparameter〉 |

〈Wertparameter〉

• Anweisungsteil:

Mindestens 1 Anweisung muß dem Funktionsnamen einen Wert zu-

weisen, z.B. mittels Zuweisungs-Anweisung:

〈Funktionsname〉 ::= 〈Ausdruck〉;

Der Funktionsname darf nicht wie eine übliche Variable benutzt wer-

den; er darf nur links bei Zuweisungen stehen.

Beispiel 4 Dekadischer Logarithmus lg x

Eine Funktion lg(x) = ln(x)/ ln(10) ist zu definieren.

function lg ( x : double) : double;

{ x - Eingangsparameter }

begin

lg := ln(x) / ln(10.0)

end;

Beispiel 5 Physikalische Kraft F (s)

Die Abhängigkeit einer Kraft F vom Weg s werde durch die folgende

reelle Funktion beschrieben:

F (s) =











e−s2 falls |s| > 1

s2/e falls 0 ≤ s ≤ 1

s+ 1 falls − 1 ≤ s < 0

function F ( s : double) : double;

{ s - Eingangsparameter }

const

e:double = 2.718281828459;

begin

if abs(s) > 1 then F := exp(-sqr(s)) else

if s >=0 then F := sqr(s) / e

else F := s+1.0

end; { of F }

Merke: Eingangsparameter werden meist als Wertparameter behandelt.

Auch Funktionen dürfen lokale Parameter enthalten. Im Deklarationsteil

des Programmes stehen Funktionen wie Prozeduren.

3.2.2 Funktions–Aufruf

Standardfunktionen werden - anders als Prozeduren - innerhalb von Aus-

drücken aufgerufen. Sie können als Operanden fungieren und liefern einen

Wert. Dieser Mechanismus gilt auch für nutzerdefinierte Funktionen. Die

Regeln für die Parameterübergabe (Variablenparameter, Wertparameter)

bei Prozeduren treffen auch auf Funktionen zu.

Beispiele für Funktions-Aufrufe:

var

t, x, y, s0, s1, s2, f1, f2 : double;

begin

readln(t,s0,s1,s2);

x := ln(t+1.0) - 2.0*lg(t+1.0);

y := exp(2.0 * lg(t + lg(x+2.0)));

{ geschachtelter Aufruf }

f1 := F(s1 - s0) - F(s2 - s1);

f2 := 0.5*( F(s0) + 2.0*F(0.5(s0+s1)) + F(s1) );

Beispiel 6 Physikalische Arbeit A

Die Abhängigkeit einer Kraft F vom Weg s werde durch die reelle Funk-

tion F (s) aus Bsp.5 dargestellt. Zu berechnen ist die physikalische

Arbeit für den Weg [a, b] mit a < b gemäß

A =

∫ b

a

F (s) ds .

Problemanalyse:

Das Integral muß i.allg. näherungsweise berechnet werden. [a, b] werde

inn Teilintervalle der Breite h = (b−a)/n unterteilt. Die Teilintegrale

A1, A2, ..., An werden durch die Trapeze T1, T2, ..., Tn angenähert.

Resultat ist die zusammengesetzte Trapezformel

T = T1 + T2 + T3 + · · · + Tn

= h ·
[

1

2
F (a) + F (a+ h) + F (a+ 2 ∗ h) + · · · +

1

2
F (b)

]

Logische Programmstruktur:

1. Hauptprogramm ’Bestimmtes_Integral’:

– Eingabe der Werte a, b, n

– Aufruf der Prozedur Trapez

– Ausgabe des Näherungswertes T des Integrals

– Wiederholung der Rechnung mit neuem n

2. Prozedur ’Trapez’:

– Berechnung von T mit y = F (x)

3. Funktion ’F’:

– Berechnung von F (s)

Physische Programmstruktur:

Program Bestimmtes_Integral;

uses Crt;

var

n : integer;

z : char;

a, b, T : double;

function F ( s : double) : double;

{ Deklaration des Integranden }

const

e:double = 2.718281828459;

begin

if abs(s) > 1 then F := exp(-sqr(s)) else

if s >=0 then F := sqr(s) / e

else F := s+1.0

end; { of F }

procedure Trapez (a, b : double; n : integer;

var T : double);

{ Bestimmtes Integral mit Trapezregel }

var

i : integer;

h,x,y0,y : double;

begin

h := (b-a)/n; x := a;

y0 := F(a); y := F(b);

T := 0.5*(y0+y);

for i:=1 to n-1 do

begin

x := a + i*h;

y := F(x);

T := T + y

end;

T := T*h

end; { of Trapez }

begin { Hauptprogramm }

ClrScr;

writeln(’Bestimmtes Integral ueber F(s)’);

writeln(’==============================’);

writeln;

write(’Untere Grenze a = ’); readln(a);

write(’Obere Grenze b = ’); readln(b);

repeat writeln;

write(’Intervallzahl n = ’); readln(n);

Trapez (a,b,n,T);

writeln(’Integralwert T = ’,T);

write(’Wiederholung (J/N)? ’); readln(z)

until upcase(z) = ’N’

end.

Bestimmtes Integral ueber F(s)

==============================

Untere Grenze a = -2

Obere Grenze b = 2

Intervallanzahl n = 500

Integralwert T = 8.876804692E-1

Wiederholung (J/N)? j

Intervallanzahl n = 1000

Integralwert T = 8.904079851E-1

Wiederholung (J/N)? n

Integral mittels Simpsonregel – Version in C++

#include <iostream.h>

#include <math.h>

double f(double x) // Funktion 1

{

return sin(x)+exp(-x);

}

double g(double x) // Funktion 2

{

return cos(2.0*x)-exp(-x);

}

// Berechnung von I aus der Simpsonregel zur Funktion f,

// dem Intervall [a,b] und der Schrittanzahl n

double simpson(double (f)(double), double a, double b,

double tol=1E-8, int n=50, int k=10)

{

if (a<=b && n>0 && !(n%2)) // Parametertest

{

double h, I, alt, neu = 1E15, abstand;

int i;

do

{

alt = neu; h = (b-a)/n; I = f(a);

for (i=1; i<n; i++)

if (i%2) // i ungerade

I = I + 4*f(a+i*h);

else // i gerade

I = I + 2*f(a+i*h);

I = h/3*(I+f(b));

neu = I;

abstand = fabs(alt-neu);

n += k;

} while (abstand >= tol);

return neu;

}

else

return 0;

}

int main() {

double a, b, tol;

cout.precision(16); // double !

cout << endl << "Integrale mit Simpsonregel" << endl;

cout << "==========================" << endl;

cout << "Linke Intervallgrenze : "; cin >> a;

cout << "Rechte Intervallgrenze: "; cin >> b;

cout << "Genauigkeitsschranke : "; cin >> tol;

cout << endl;

cout << "Gegebene Toleranz : " << endl;

cout << "Integral1 = " << simpson(f,a,b,tol) << endl;

cout << "Integral2 = " << simpson(g,a,b,tol) << endl;

cout << endl;

cout << "Standard-Toleranz : " << endl;

cout << "Integral1 = " << simpson(f,a,b) << endl;

cout << "Integral2 = " << simpson(g,a,b) << endl;

return 0;

}

3.3 Blockstruktur und globale Objekte

Ein Block ist gemäß Syntax der Hauptteil eines Programmes, einer Pro-

zedur oder einer Funktion. In BP werden auch Units als selbständige

Blöcke betrachtet.

• Externe Blöcke – selbständige, von anderen Blöcken weitgehend un-

abhängige Programmteile, die übersetzt werden können - und nicht

in anderen Blöcken enthalten sind (BP: Program, Unit)

• Interne Blöcke – in übergeordneten (internen oder externen) Blöcken

enthaltene Programmteile, die nicht selbständig übersetzt werden

können (BP: Procedure, Function)

3.3.1 Blockniveau, Lebensdauer und Gültigkeitsbereich

PASCAL gestattet die Deklaration von Blöcken im Innern anderer Blöcke

(“geschachtelte Prozeduren“).

Blockniveau:

Es gibt die Anzahl der Programme, Units, Funktionen, Prozeduren an, in

denen die jeweilige Prozedur-/Funktions-Deklaration enthalten ist, d.h.

deren Schachtelungstiefe.

Blockniveaus sind 0 (externe Blöcke), 1, 2, 3,... (interne Blöcke).

Beispiel 7 Blockniveaus (vgl. nächste Seite)

• HP besitzt Blockniveau 0,

• UP1 und UP2 sind von Niveau 1 und

• UP2 besitzt Blockniveau 2.

Program HP;

var j,y : integer; . . .

Procedure UP1;

var i,j : integer; . . .

Procedure UP2;

var y,j : integer; . . .

begin

j := 10; y := 1; . . .

end; { of UP2 }

begin

UP2;

i := 15; j := 20;

y := 2; . . .

end; { of UP1 }

Procedure UP3;

var i,j : integer; . . .

begin

j := 50; . . .

end; { of UP3 }

begin

UP1; { falsch: UP2; }

j := 40;. . .

UP3;

end. { of HP }

• Rationelle Speichernutzung:

Zielsetzung der Blockstrukturierung: Allen Programmobjekten (Kon-

stanten, Variablen, Prozeduren, Funktionen etc.) ist nur solange

Speicherplatz zuzuweisen, wie sie gebraucht werden!

• Lebensdauer eines Objektes:

Sie definiert den Zeitraum, in dem deklarierten Bezeichnern im Spei-

cher reale, physikalische Objekte zugewiesen sind. Bei den Compi-

lierungszeit-Objekten unterscheidet man

– statische (globale) Objekte: Zuweisung während der gesamten

Programmdauer – in festen Datensegmenten; betrifft: Variablen,

Konstanten auf Niveau 0, Funktionen und Prozeduren

– automatische (lokale) Objekte: Zuweisung bei Eintritt in einen

internen Block und (“automatischer“) Verlust des Speicherplat-

zes bei Verlassen dieses Blockes – im Stack (Stapel, Kellerspei-

cher); betrifft: Variablen, Konstanten auf Niveau 1,2,3,...

Beispiel 7 (s.o)

– Statische (globale) Objekte: j,y,UP1,UP2,UP3

– Automatische (lokale) Objekte: i,j (in UP1);

y,j (in UP2); i,j (in UP3)

Laufzeit-Objekte mit dynamischer Lebensdauer werden später behandelt.

• Gültigkeitsbereich G:

Der GültigkeitsbereichG eines Bezeichners ist derjenige Programm-

teil, in dem über den Bezeichner auf das Objekt zugegriffen werden

kann. Für statische und automatische Objekte ist G genau derje-

nige Block B, in dem der zugehörige Bezeichner deklariert ist -

beginnend mit der Deklaration.

Beispiel 7 (s.o) Darstellung der Gültigkeitsbereiche:

– Statische (globale) Objekte: j,y,UP1,UP2,P3

– Automatische (lokale) Objekte: i,j (in UP1);

y,j (in UP2); i,j (in UP3)

• Sichtbarkeit:

Unter der Sichtbarkeit eines Bezeichners versteht man denjenigen

Teil seines Gültigkeitsbereiches, von dem aus ein zulässiger Zugriff

auf das Objekt möglich ist. Das Auftreten eines doppelten Bezeich-

ners unterbricht die Sichtbarkeit, die erst wieder eintritt, wenn der

Gültigkeitsbereich des doppelten Bezeichners endet.

Beispiel 7 (s.o) Sichtbarkeit von j,y,i

Beispiel 7 (s.o.) Darstellung der

• Gültigkeits- und Sichtbarkeitsbereiche

• Speicherbelegung von Datensegment und Stack

3.3.2 Globale Programmobjekte

B sei der Block eines Programmes, einer Prozedur oder einer Funktion.

Ein Programmobjekt x mit Gültigkeitsbereich G heißt

• lokal bezüglich B, falls G ⊆ B

• global bezüglich B, falls G ⊃ B

gilt. Im 2.Fall wird x außerhalb von Block B deklariert.

Beispiel 7 (s.o)

Variable y von HP ist global bzgl. UP1, UP2 und UP3 ⇒ Anweisung

y := 2; ist zulässig.

Empfehlungen:

• Die Benutzung globaler Objekte innerhalb von Prozeduren / Funk-

tionen sollte in Interesse der Klarheit und Universalität weitgehend

vermieden werden.

• Mitunter macht der - sparsame - Einsatz globaler Konstanten, Va-

riablen oder Funktionen jedoch Sinn. Bei globalen Variablen sollte

eine wertmäßige Veränderung im internen Block unbedingt unter-

bleiben.

Beispiel 8 Masse eines inhomogenen Stabes

Die Abhängigkeit der Dichte ̺(s) von der Koordinate s werde durch

̺(s) =

{

̺A · e−(s−0.7)2 falls 0 < s ≤ 1.5

̺B + 0.11s falls 1.5 ≤ s ≤ 5.0

dargestellt.

Skizze:

Die Abhängigkeit der Dichte ̺(s) von der Koordinate s werde durch

̺(s) =

{

̺A · e−(s−0.7)2 falls 0 < s ≤ 1.5

̺B + 0.11s falls 1.5 ≤ s ≤ 5.0

dargestellt. ̺A und ̺B sind gegebene konstante Dichten der 2 Stab-

komponenten. Der Stabquerschnitt sei mit A = 1 als konstant ange-

nommen.

Zu berechnen ist die Masse über [0, L] mit Stablänge L ≤ 5.0 gemäß

m =

∫ L

0

A̺(s) ds .

Problemanalyse:

1. Das Integral wird näherungsweise mittels der Trapezformel (Bsp. 6)

mit n Teilintervallen berechnet.

2. ̺A und ̺B werden als globale Konstanten deklariert; die Stablänge

L und die Zahl n der Teilintervalle sollen variabel sein.

Logische Programmstruktur:

1. Hauptprogramm ’Stabmasse’:

– Eingabe der Werte L, n

– Aufruf der Prozedur Trapez

– Ausgabe des Näherungswertes m des Integrals

– Wiederholung der Rechnung mit neuen L, n

2. Prozedur ’Trapez’:

– Berechnung von T mit y = F (x)

3. Funktion ’F’:

– Berechnung der Dichtefunktion F (s)

Physische Programmstruktur:

Program Stabmasse;

uses Crt;

const { Materialdichten }

rho_A:double = 3.658;

rho_B:double = 2.045;

var

n : integer;

z : char;

L, m : double;

function F ( s : double) : double;

{ Deklaration des Integranden }

begin

if (0 < s) and (s< 1.5) then

F := rho_A * exp(-sqr(s-0.7)) else

F := rho_B + 0.11 * s

end; { of F }

procedure Trapez (a, b : double;

n : integer; var T : double);

{ Bestimmtes Integral mit Trapezregel }

{ vgl. Beispiel 6 }

. . .

y0 := F(a); y := F(b);

. . .

y := F(x);

. . .

end; { of Trapez }

begin { Hauptprogramm }

ClrScr;

writeln(’Masse m eines Stabes’);

writeln(’====================’);

repeat writeln;

write (’Stablaenge L = ’); readln(L);

write (’Intervallzahl n = ’); readln(n);

Trapez (0,L,n,m);

writeln(’Gesamtmasse m = ’, m:12:6);

write (’Wiederholung (J/N)? ’); readln(z)

until upcase(z) = ’N’

end.

Masse eines Stabes

==================

Stablaenge L = 4.6

Intervallanzahl n = 100

Gesamtmasse m = 11.979221

Wiederholung (J/N)? j

Stablaenge L = 4.6

Intervallanzahl n = 200

Gesamtmasse m = 11.978402

Wiederholung (J/N)? j

Stablaenge L = 2.9

Intervallanzahl n = 500

Gesamtmasse m = 7.804488

Wiederholung (J/N)? n

Wertung:

(a) Die Konstanten rho_A, rho_B sind global bzgl. der Funktion F.

(b) Die Funktion F ist global bezüglich der Prozedur Trapez.



Kapitel 4

Grundlegende Datenstrukturen

Felder (array) und Zeichenketten (string) sind die häufigsten Datenstruk-

turen im Bereich mathematisch-naturwissenschaftlich-technischer Algo-

rithmen. Charakteristisch ist die Festlegung der Strukur dieser Daten -

einschließlich des Speicherbedarfs - während der Compilierungszeit (sta-

tische bzw. automatische Speicherklasse). Syntaktisch werden diese und

andere Strukturen in PASCAL vermittels Datentypen definiert.

4.1 Typdefinitionen

Eine Datentyp-Definition legt die Menge der Werte fest, die die Varia-

blen des entsprechenden Typs annehmen können und ordnet dem Typ

einen Bezeichner (Typnamen) zu. Damit kann der Programmierer neben

oder mit den eingeführten Standardtypen (integer, longint, real, dou-

ble,...) neue eigene “abstrakte“ Datentypen erzeugen. Dies geschieht im

Deklarationsteil des Programmes, der Prozedur oder der Funktion.

Objekt Deklarationsart Res. Wort

Variable Variablendeklaration Var

Konstante Konstantendeklaration Const

Datentyp Typdeklaration Type

Funktion Funktionsdeklaration Function

Verfahren Prozedurdeklaration Procedure

Syntax:

〈Typ-Deklaration〉 ::= Type 〈Typ-Definition〉{; 〈Typ-Definition〉};

〈Typ-Definition〉 ::= 〈Bezeichner〉 = 〈Typ〉
〈Typ〉 ::= 〈einfacher Typ〉|〈strukturierter Typ〉|〈Zeigertyp〉

〈einfacher Typ〉 ::= 〈Aufzählungstyp〉|〈Teilbereichstyp〉|〈Typbezeichner〉
Aufzählungstypen:

Ein Aufzählungstyp definiert eine geordnete Menge von Werten durch

Aufzählung der Bezeichner, die diese Werte ausdrücken. Die Definition

gibt - in linearer Ordnung - alle möglichen Werte (d.h. die Konstanten)

an.

Syntax: 〈Aufzählungstyp〉 ::= (〈Bezeichner〉 {, 〈Bezeichner〉 })

Beispiele:

Type Tag = (MO,DI,MI,DO,FR,SA,SO);

geschlecht = (weiblich, maennlich);

Richtung = (Sueden, Westen, Norden, Osten);

Marke = (audi,bmw,citroen,ferrari,suzuki,vw);

Studiengang = (Maschinenbau, Mechatronik);

Var Feiertag, Pruefungstag : Tag;

f,g : geschlecht;

Neuwagen, Rostlaube : Marke;

Fachrichtung : Studiengang;

Begin Pruefungstag := MI;

Fachrichtung := Mechatronik;

Neuwagen := ferrari;

Rostlaube := Neuwagen;

if Pruefungstag < FR then Feiertag := FR;

. . .

Merke: Bei Aufzählungstypen sind Wertzuweisungen und Vergleiche (ent-

sprechend der Aufzählungs-Reihenfolge) möglich. Für die skalaren Stan-

dardtypen Char und Boolean gilt:

(a) ... < ’A’ < ’B’ < ... < ’Z’ < ’a’ < ... < ’z’ < ...

(b) Type Boolean = (false, true); mit false < true

Teilbereichstypen:

Ein Datentyp T ist Teilbereich eines anderen, wenn die Menge seiner

möglichen Werte in der Menge der möglichen Werte des umfassenderen

Typs enthalten ist. Man benutzt dazu i.allg. die Teilbereichs-Notation.

Syntax: 〈Teilbereichstyp〉 ::= 〈Konstante〉 .. 〈Konstante〉
Beispiele:

Type Tag = (MO,DI,MI,DO,FR,SA,SO);

Werktag = MO .. FR;

Ziffer = 0 .. 9; { Grundtyp: }

Winkel = 0 .. 360; { integer }

Grossbuchstabe = ’A’ .. ’Z’; { Grundtyp: }

Kleinbuchstabe = ’a’ .. ’z’; { char }

Var Feiertag : Tag;

Pruefungstag : Werktag;

alpha1, psi : Winkel;

. . .

Merke: Beide Konstanten in der Definition müssen in aufsteigender An-

ordnung im Grundtyp stehen. Der Grundtyp muß zuvor definiert worden

sein oder ist ein Standardtyp. Von float-Typen sind keine Teilbereiche

möglich!

Typbezeichner:

Hiermit findet lediglich eine Neubezeichnung eines Standardtyps oder

eines bereits definierten Typs statt (Synonymgebung, Alias-Name).

Beispiele:

Type ganz = integer;

logical = Boolean;

float = double;

int1 = byte;

int4 = longint;

character = Char;

Var n,k : ganz;

p,q,r : logical;

z1, z2 : character;

Standardfunktionen für skalare Typen:

1. succ(x) bestimmt den Nachfolger in der Anordnung

2. pred(x) bestimmt den Vorgänger in der Anordnung

3. ord(x) bestimmt den Ordinalwert in der Anordnung

Merke: Das 1. Element der Aufzählung hat die Platznummer 0.

Beispiele:

Type Tag = (MO,DI,MI,DO,FR,SA,SO);

succ(DO) -> FR succ(12) -> 13 ord(false) -> 0

pred(MI) -> DI pred(-3) -> -4 ord(true) -> 1

ord (DI) -> 1 succ(’B’) -> ’C’

Die strukturierten Typen array, string, record, file werden nun behandelt; Zeigertypen dagegen erst

zum späteren Zeitpunkt in Kapitel 9.

4.2 Algorithmen auf Feldern (array-Typ)

Definition 1 : Feld (array)

• Ein Feld ist eine aus einer vorher festgelegten Anzahl von Kompo-

nenten gleichen Typs bestehende Datenstruktur.

• Zugriff auf eine Komponente des Feldes erfolgt durch Angabe von

Indizes.

• Die Anzahl der Indizes zur Auswahl einer Komponente gibt die

Dimension des Feldes an. Diese kann beliebig groß sein.

4.2.1 Felddeklaration und -speicherung

Sie erfolgt am besten in Form einer Typdefinition im Deklarationsteil.

Der Feldtyp erhält dadurch einen Bezeichner zugeordnet, der in seinem

Gültigkeitsbereich zur Deklaration von Variablen und Prozedurparame-

tern benutzt werden kann.

Syntax:

〈Typ-Definition〉 ::= 〈Feldtyp〉 = array[〈Indextyp〉{, 〈Indextyp〉}]

of 〈Komponententyp〉
〈Feldtyp〉 ::= 〈Bezeichner〉

Semantik:

• Der Indextyp muß ein Aufzählungstyp sein.

• Die Zahl der Indextypen ergibt die Dimension n des Feldtyps.

• Der Komponententyp kann beliebig sein.

Beispiele:

{1) Numerische Felder }

type Zahlen = array [1 .. 20] of double;

var a, b : Zahlen;

{2) Nicht-numerische Felder }

type Kette = array [1..80] of char;

Tag = (MO,DI,MI,DO,FR,SA,SO);

Woche = array [1..7] of Tag;

var Text1, Text2 : Kette;

w : Woche;

{3) Mehrdimensionale Felder }

const n = 20;

m = 30;

type Matrix = array [0..n, 1..m] of double;

Tafel = array [Boolean,Boolean,Boolean]

of Boolean;

var A, B : Matrix;

Wert : Tafel;

Speicherung eindimensionaler Felder:

Eindimensionale Felder: werden linear in Reihenfolge der Komponenten

abgespeichert, indem der Indextyp alle Werte des Indexbereichs durchläuft

Beispiel:

type Tag = (MO,DI,MI,DO,FR,SA,SO);

Woche = array [1..7] of Tag;

var w : Woche;

DI FR FR MO ⇐= Komponenten

1 2 3 4 5 6 7 ⇐= Indexbereich (IB)

Speicherung mehrdimensionaler Felder:

Ist der Komponententyp eines Feldtyps wiederum ein Feldtyp, so liegt

ein mehrdimensionales Feld vor.

Beispiel:

type Zeile = array [1..4] of integer;

Matrix = array [1..3] of Zeile;

var M : Matrix;

2 -3 0 0 0 0 1 1 ⇐= Komp.

1 2 3 4 1 2 3 4 1 2 3 4 ⇐= IB Zeile

1 2 3 ⇐= IB Matrix

Äquivalent sind die Typdeklarationen

type Matrix = array [1..3] of array [1..4] of integer;

type Matrix = array [1..3, 1..4] of integer;

Mehrdimensionale Felder werden allgemein in lexikografischer Reihenfol-

ge abgespeichert, d.h. die Komponenten sind linear so angeordnet, daß

der letzte Index am schnellsten variiert, der erste Index dagegen am lang-

samsten.

Insbesondere werden Matrizen zeilenweise gespeichert!

Indizierte Variablen:

Der Zugriff auf eine Feldkomponente erfolgt durch Angabe der dafür

nötigen Indizes.

Syntax: 〈Feldvariable〉 [〈Indexausdruck〉 {, 〈Indexausdruck〉} ]

Beispiele:

const n = 20; m = 30;

type Zahlen = array [1 .. 20] of double;

Matrix = array [0..n, 1..m] of double;

Tensor = array [1..2, 1..2, 1..2] of single;

var x, y : Zahlen;

A, B : Matrix;

V : Tensor;

i,j,k : integer;

{ Indizierte Variablen }

x[2], A[1,4], V[1,2,2],

y[i+1], B[i-1-k, 2*k+1], V[i,j,k+i]

Äquivalent sind die Variablenzugriffe

(i) V [i, j, k]

(ii) V [i][j, k]

(iii) V [i][j][k]

Felder als Prozedurparameter:

Strukturierte Datentypen dürfen nicht in Formalparameter-Abschnitten

stehen. Ausweg: Deklaration eines Typbezeichners im aufrufenden Pro-

gramm!

Beispiel 1 Skalarprodukt zweier Vektoren

Von gegebenen Vektoren a und b mit n Komponenten (n ≤ 200)
ist das Skalarprodukt

s = skalar(a, b) =
n
∑

k=1

ak · bk , ak, bk ∈ R

mit einer Funktion skalar zu berechnen.

Program main;

type Vektor = array [1..200] of double;

function skalar (n : integer;

var a,b : Vektor) : double;

var i : integer;

sum : double;

begin

sum := 0;

for i:= 1 to n do

sum := sum + a[i] * b[i];

skalar := sum

end; { of skalar }

. . .

end.

4.2.2 Numerische Feldverarbeitung

Wegen der großen Bedeutung ein- und zweidimensionaler Zahlenfelder

in der Technik werden 2 typische Grundalgorithmen - in Form von Pro-

zeduren - betrachtet.

Beispiel 2 Methode der kleinsten Quadrate

Gegeben sind n Meßwerte (xi, yi), i = 1(1)n (z.B. x - Zeit, y

- Temperatur), zwischen denen ein linearer Zusammenhang vermutet

wird:

y = f(x) = ax+ b

Man bestimme die Koeffizienten a und b so, daß die Gesamtsumme der

Fehler minimal wird!

Problemanalyse:

• Quadratsumme der Fehler

Q = Q(a, b) =
n
∑

i=1

[f(xi) − yi]
2 −→ Min!

• Methode der kleinsten Quadrate (C.F.Gauss, 1809):

∂Q

∂b
= 2

n
∑

i=1

[axi + b− yi] = 0

∂Q

∂a
= 2

n
∑

i=1

[axi + b− yi]xi = 0

• Normal(en)gleichungen:

n · b+

(

n
∑

i=1

xi

)

· a =
n
∑

i=1

yi

(

n
∑

i=1

xi

)

· b+

(

n
∑

i=1

x2
i

)

· a =
n
∑

i=1

xiyi

• Hilfsgrößen:

sx :=
n
∑

i=1

xi, sy :=
n
∑

i=1

yi, sxx :=
n
∑

i=1

x2
i ,

syy :=
n
∑

i=1

y2
i , sxy :=

n
∑

i=1

xiyi

• Cramersche Regel:

D := n · sxx− (sx)2

b := (sy · sxx− sx · sxy)/D
a := (n · sxy − sx · sy)/D

{ Deklarationen im Hauptprogramm }

const nmax = 1000; { max. Messwertezahl }

type Vektor = array [1 .. nmax] of double;

procedure Least_squares (

n : integer; { Anzahl }

var x, y : Vektor; { Messwerte }

var a, b : double; { Koeffizienten }

var Q : double ); {Quadratsumme }

var D,hx,hy,sx,sy,sxx,syy,sxy : double;

i : integer;

begin

sx :=0; sy :=0; sxx :=0; syy :=0; sxy :=0;

for i:= 1 to n do

begin

hx := x[i]; hy := y[i];

sx := sx + hx;

sy := sy + hy;

sxx := sxx +hx * hx;

syy := syy +hy * hy;

sxy := sxy +hx * hy

end;

D := n * sxx - sx * sx;

b := (sy * sxx - sx * sxy) / D;

a := (n * sxy - sx * sy ) / D;

Q := a*a*sxx + b*b*n + syy +

(a*b*sx - a*sxy - b*sy) * 2.0

end; { of Least_squares }

Beispiel 3 Lineare Gleichungssysteme

Mechanische Schwingungsprobleme sowie die Analyse elektrischer Netz-

werke erfordern die Lösung großer linearer Gleichungssysteme mit n Glei-

chungen für n Unbekannte x1, x2, ..., xn.

• Allgemeine Darstellung:

a11x1 + a12x2 + · · · + a1nxn = a1

a21x1 + a22x2 + · · · + a2nxn = a2
... ... ... ...

an1x1 + an2x2 + · · · + annxn = an

bzw. in Matrixform Ax = a mit

A =











a11 a12 · · · a1n

a21 a22 · · · a2n

. . . . . . . . . . . . . . . . . .

an1 an2 · · · ann











, a =











a1

a2
...

an











, x =











x1

x2
...

xn











• Verketteter Gaußscher Algorithmus:

– Überführung des Systems in ein gestaffeltes Dreiecksystem mit-

tels LR-Zerlegung in einem einzigen Arbeitsgang, der alle einzel-

nen Eliminationsschritte miteinander verknüpft (“verkettet”)

r11x1 + r12x2 + · · · + r1nxn = r1

r22x2 + · · · + r2nxn = r2

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .

rnnxn = rn

– Falls erforderlich, erfolgen Zeilenvertauschungen. Ziel: rii 6= 0

für alle i (Falls A regulär ist, so ist das stets erreichbar!)

– Rekursive Berechnung der Lösungen xn, xn−1, ..., x2, x1 aus
dem gestaffelten System:

xi = (ri − xnrin − ...− xi+1ri,i+1)/rii , i = n(−1)1

Prozedurparameter:

n – Anzahl der Unbekannten (n ≤ nmax ≤ 1000)

a – Koeffizientenmatrix mit aik reell

b – Vektor der rechten Seiten bi reell

x – Vektor der Lösungen xi reell

{ Deklarationen des Hauptprogramms }

const maxn = 100; { Max. Zahl der Unbekannten }

type vektor = array [1..maxn] of double;

matrix = array [1..maxn,1..maxn] of double;

procedure Gauss(

n : byte; { Zahl der Unbekannten }

var a : matrix; { Koeffiizientenmatrix }

var b : vektor; { Rechte Seiten }

var x : vektor; { Loesungen }

var sing : boolean); { Singularitaet }

{ *** Loesung des linearen Gleichungssystems Ax=b

mit dem verketteten GAUSSschen Algorithmus

(LR-Zerlegung) mit Zeilenpivotisierung ****** }

label M1; { Marke fuer singulaeren Fall }

const eps = 1e-15; { Abbruchgenauigkeit }

var max, s : double;

i, j, m, index : byte;

begin

sing:=false; {Annahme: Matrix A ist regulaer }

for m:=1 to n do { LR-Zerlegung }

begin { Pivotsuche }

max:=0;

for i:=m to n do

begin s:=a[i,m];

for j:=1 to m-1 do s:=s+a[i,j]*a[j,m];

a[i,m]:=s;

if abs(s) > abs(max) then

begin max:=s; index:=i end

end;

{ Test auf Singularitaet}

if abs(max) < eps then

begin sing:=true; goto M1 end;

if index > m then { Zeilenvertauschung }

begin

for i:=1 to n do

begin s:=a[m,i]; a[m,i]:=a[index,i];

a[index,i]:=s

end;

s:=b[m]; b[m]:=b[index]; b[index]:=s

end;

{ Bestimmung der Spaltenelemente }

for i:=m+1 to n do a[i,m]:=-a[i,m]/max;

{ Bestimmung der Zeilenelemente }

for i:=m+1 to n do

begin s:=a[m,i];

for j:=1 to m-1 do s:=s+a[m,j]*a[j,i];

a[m,i]:=s

end;

s:=b[m];

for j:=1 to m-1 do s:=s+a[m,j]*b[j];

b[m]:=s

end; { of LR-Zerlegung}

{ Rueckwaerts-Elimination }

for i:=n downto 1 do

begin s:=-b[i];

for j:=n downto i+1 do s:=s+a[i,j]*x[j];

x[i]:=-s/a[i,i]

end; { of Rueckwaerts-Elimination }

M1:

end; { of Gauss }

Dynamische Felder:

Alle eingeführten arrays sind Compilierungszeit-Objekte; die Indexgren-

zen müssen also bereits während der Complilierung berechenbar sein.

Wird die Speicherfestlegung während der Laufzeit angestrebt, so sind

dynamische Felder zu benutzen. Der Speicherplatz kann dann zur Lauf-

zeit explizit reserviert und wieder freigegeben werden (Laufzeit-Objekte).

Das eleganteste und bequemste Konzept bietet GPC mit den Schema-

Typen (vgl. Kap. 4.4).

4.3 Algorithmen auf Zeichenketten (string-Typ)

Zeichenketten können als array[ ] of char mit fester Länge de-

finiert werden. Um jedoch Zeichenketten variabler Länge verarbeiten zu

können, existieren in BP

• Zeichenketten des Typs string variabler (dynamischer) Länge

mit Maximallänge 255

• Null-terminierte Zeichenketten (theoretisch) beliebiger Länge.

4.3.1 Zeichenketten-Definition und -Verarbeitung

Abspeicherung von Zeichenketten:

• Jede Zeichenkette besitzt eine vereinbarte maximale Länge (Zei-

chenzahl) nmax und eine aktuelle Länge n , wobei n im Bereich

0 ≤ n ≤ nmax variiert.
• Eine Zeichenkette belegt nmax+ 1 Bytes:

0. Byte : belegt durch aktuelle Länge (LI)
1. Byte : belegt durch 1. Zeichen
2. Byte : belegt durch 2. Zeichen
. . . : . . . . . . . . . . . .
n-tes Byte : belegt durch n-tes Zeichen
(n+1)-tes Byte : unbestimmt
. . . : . . . . . . . . .
nmax-tes Byte : unbestimmt.

• Folgerung für nmax: 1 ≤ nmax ≤ 255

• Skizze zur Abspeicherung:

0 1 2 n-1 n n+1 nmax

n T U . . . a u . . .

Typ- und Variablendefinition

Syntax: 〈Kettentyp〉 ::= string | string [〈Kettenlänge〉]
〈Kettenlänge〉 ::= 〈integer-Konstante〉

Kettenkonstante ::= ′′ | ′〈Zeichen〉{〈Zeichen〉}′

Semantik:

• Fehlt die Kettenlänge, so wird die Maximallänge 255 angenommen;

andernfalls muß 1 ≤ Kettenlänge ≤ 255 gelten.

• In BP existiert auch die leere Zeichenkette.

• Variablendeklarationen erfolgen in üblicher Weise:

Var name_1, name_2, ... , name_N : Kettentyp;

Beispiele:

{1) Vereinbarungen }

const laenge = 80;

type Namen = string[25];

infos = string[20];

zeile = string[laenge];

var Vorname, Zuname : Namen;

Bildschirm : zeile;

uni, titel, ort : infos;

{2) Zuweisungen von Konstanten}

Vorname := ’Harry’;

Zuname := ’Potter’;

titel := ’TU Ilmenau’;

uni := titel;

titel := ’’;

Zugriff auf einzelne Zeichen:

Die Zeichen einer Kette sind mittels indizierter Variablen

〈Name〉[〈Indexausdruck〉]

erreichbar. Das Ergebnis kann einer Variablen des Typs char zuge-

wiesen werden. Die Komponente 0 liefert den Längenindikator (LI) - als

Zeichen des Typs char.

Beispiele:

var ch : char;

n : integer;

begin

uni := ’TU Ilmenau’;

titel := ’’;

ch := uni[2]; --> ’U’

ch := uni[0]; --> Steuerzeichen

n := ord(uni[0]); --> 10

n := ord(titel[0]); --> 0

Kettenausdrücke und - anweisungen

1. Kettenzuweisung:

〈Kettenvariable〉 := 〈Kettenausdruck〉

Einfachste Ausdrücke sind Kettenkonstanten und -variablen. Ist der

LI des Ausdruckes größer als die Maximallänge der Kettenvariablen,

so wird rechts abgeschnitten.

2. Verkettung (Konkatenation):

Durch den Operator + werden 2 Ketten aneinandergehängt (verket-

tet). Dabei sind die aktuellen Längen n1, n2 der beteiligten Ketten

maßgeblich! n = n1 + n2 ist die Länge der neuen Zeichenkette.

Beispiele:

{1) Vereinbarungen }

type string20 = string[20];

string15 = string[15];

var name, titel : string20;

ort : string15;

adresse : string;

{2) Zuweisungen von Konstanten}

name := ’TU Ilmenau, Ehrenberg’;

ort := name; { --> Informationsverlust! }

{3) Verkettung}

name := ’TU Ilmenau, Ehrenberg’;

ort := ’98684 Ilmenau’;

adresse := name + ort; { --> Abstaende! }

adresse := name + ’, ’ + ort;

Zeichenketten-Vergleiche:

• Die Priorität der Relationsoperatoren =, <>,<,>,<=, >=

ist geringer als die der Verkettung + .

• Ketten werden zeichenweise von links nach rechts gemäß ASCII-

Code verglichen. Sie sind nur dann gleich, wenn Inhalt und aktuelle

Länge übereinstimmen.

• Haben 2 Ketten verschiedene Länge, stimmen jedoch bis zum letzten

zeichen der kürzeren Kette überein, so ist die kürzere Kette kleiner.

• Der Ergebniswert ist vom Typ Boolean.

• Beispiel: < und ≤ gilt für die Anordnung

Mai < Maier < Mayer < Meier < Meier1 <

< Meyer < Meyer-Hoff < MeyerHoff < Meyerhoff



Teil1.tex ♣ Doz.Dr.habil.Werner Vogt ♠ Algorithmen und Programmierung ♣ WS 2005 ♠ 14. September 2005 105

Beispiel 4 Schnelles Sortieren von Namen

Nach kompletter Eingabe von maximal 1000 Namen sind diese lexikogra-

fisch zu sortieren und anschließend in sortierter Reihenfolge auszugeben.

Program Sortierung_von_Namen_durch_Partitionierung;

uses crt;

const maxlaenge = 25; { max. Namenslaenge }

maxanzahl = 1000; { max. Namensanzahl }

type element = string[maxlaenge];

vektor = array[1..maxanzahl] of element;

var i,n : integer; a : vektor;

procedure QUICKSORT(var a:vektor; l,r:integer);

{ Sortieren der Elemente a[l],a[l+1],...,a[r]

in aufsteigender Anordnung }

var i,j : integer; g : element;

begin

i := l; j := r; g := a[i];

{ Divide ! ... -> Teilen des Feldes }

repeat

while (a[j] >= g) and (i < j) do dec(j);

a[i] := a[j];

while (a[i] <= g) and (i < j) do inc(i);

a[j] := a[i]

until i = j;

a[i] := g;

{ ... et impera ! -> Sortieren der Teile }

if l < i-1 then QUICKSORT(a,l,i-1);

if i+1 < r then QUICKSORT(a,i+1,r)

end; { of QUICKSORT }

begin { Hauptprogramm }

ClrScr;

writeln(’Sortieren mit Quicksort’);

writeln(’=======================’);

writeln;

write(’Anzahl = ’); readln(n); writeln;

writeln(’Unsortierte Folge :’);

for i := 1 to n do

begin

write(’Name[’,i,’] = ’); readln(a[i])

end;

writeln;

QUICKSORT(a,1,n);

writeln(’Sortierte Folge :’);

for i := 1 to n do

writeln(’Name[’,i,’] = ’,a[i]);

readln

end.

4.3.2 Zeichenketten-Prozeduren

Zur Vereinfachung der Zeichenketten-Verarbeitung stehen in BP folgen-

de Prozeduren zur Verfügung.

• Löschen in einer Zeichenkette

Format : Delete (St, Pos, Anzahl);

Parameter : St – Bezeichner einer Kettenvariablen

Pos – integer-Ausdruck

Anzahl – integer-Ausdruck

Wirkung : Aus Kette St werden Anzahl Zeichen gelöscht,

beginnend mit Position Pos.

• Einfügen in eine Zeichenkette

Format : Insert (Objekt, St, Pos);

Parameter : Objekt – Kettenausdruck

St – Kettenvariable

Pos – integer-Ausdruck

Wirkung : Einfügen der Zeichenkette Objekt in die

Kette St ab Position Pos.

• Umwandlung in eine Zeichenkette

Format : Str (Par, St);

Parameter : Par – Schreibparameter in 3 Formen:

(a) Ausdruck

(a) Ausdruck : Format1

(a) Ausdruck : Format1 : Format2

St – Kettenvariable

Wirkung : Umwandlung des numerischen Wertes Par in eine

Zeichenkette mit Bezeichner St. Str wirkt wie

write, das Ergebnis wird St zugeordnet.

• Umwandlung in einen numerischen Wert

Format : Val (St, Var, Code);

Parameter : St – Zeichenkettenausdruck, der einen

numerischen Wert ausdrückt

Var – integer- oder real-Variable

Code – integer-Variable

Wirkung : Umwandlung des Kettenausdruckes St in einen

real- oder integer-Wert und Zuordnung zu Var.

Code = 0 , falls die Umwandlung erfolgreich war,

andernfalls gibt Code die Position des ersten

fehlerhaften Zeichens aus St an.

Beispiel 5 Zahleneingabe mit Fehlerkontrolle

Beim Einlesen von double-Zahlen sind Eingabefehler (syntaktisch) fest-

zustellen.

procedure double_input (var a : double);

{ Eingabe von double-Zahlen mit Syntaxtest }

var zk : string[40];

c : integer;

begin

readln(zk); { Eingabe als Zeichenkette }

val (zk, a, c); { Konvertierungs-Versuch }

while c <> 0 do { Wiederholungen }

begin

write(’Syntaxfehler! Neue Eingabe : ’);

readln(zk);

val (zk, a, c);

end; { of while }

end; { of double_input }

4.3.3 Zeichenketten-Funktionen

Wie auch Standardfunktionen dürfen sie – anders als die Zeichenketten-

Prozeduren – direkt in Ausdrücken benutzt werden.

• Länge einer Zeichenkette

Format : Length (St)

Parameter : St – Zeichenkettenausdruck

Wirkung : Aktuelle Länge des Kettenausdruckes St.

Ergebnistyp : integer

• Erzeugung einer Teilkette

Format : Copy (St, Pos, Anzahl)

Parameter : St – Zeichenkettenausdruck

Pos, Anzahl – integer-Ausdrücke

Wirkung : Bildung einer Teilkette aus St mit Anzahl

Zeichen, beginnend mit der Position Pos.

Ergebnistyp : string-Typ

• Suchen einer Teilkette

Format : Pos (Objekt, St)

Parameter : Objekt, St – Zeichenkettenausdrücke

Wirkung : Durchsuchen der Kette St nach dem ersten Vor-

kommen des Ausdruckes Objekt. Pos liefert

diejenige Position in St, die das erste Zeichen

von Objekt innehat (Andernfalls liefert Pos: 0).

Ergebnistyp : integer

Beispiel 6 Textanalyse

Es ist festzustellen, wie oft eine bestimmte Zeichenfolge z der maximalen

Länge 50 in einem gegebenen Text zk enthalten ist.

Beispiele:

zk := ’Pfeiffer’; z := ’f’; --> 3-mal

zk := ’Begin x := 2.3; y := x*x;

zk := ’’Pfeiffer’’; z := ’’f’’

End.’; z := ’:=’; --> 4-mal

{ Deklarationen im aufrufenden Programm }

type string50 = string[50];

function Anzahl ( zk : string; { Gegebene Kette }

z : string50) { Suchkette }

: integer; { Anzahl }

var a, p, l : integer;

y : string;

begin

l := length (z) -1; { Hilfsgroesse }

a := 0; { Zaehler fuer Anzahl }

y := zk; { Kopie der gegebenen Kettte }

p := Pos(z,y); { 1.Position von z in y }

while p <> 0 do begin

inc(a);

delete(y,1,p+l); { Abschneiden }

p := Pos(z,y) { 1.Position von z in y }

end; {of while }

Anzahl := a

end; {of Anzahl }

4.4 Schema-Typen in GNU Pascal (GPC)

Schema–Typen sind als Erweiterung im Extended Pascal (ISO 10206)

auch in GNU Pascal implementiert. Sie sind in der Regel ein- und mehr-

dimensionale Felder, die von
”
Diskriminanten“ abhängen. Diesen Varia-

blen können

– während der Compilierung oder

– während der Laufzeit

Werte zugeordnet werden.

Beispiele mit festen Diskriminantenwerten:

type

vektor(n : integer) = array[1..n] of double;

kette(maxi : integer) = array[1..maxi] of char;

matrix(m,n : integer) = array[1..m,1..n] of double;

{ Diskriminanten sind n, maxi, m,n }

var { 3 Schema-Variablen }

x : vektor(450);

a : matrix(1000,1000);

W : kette(200000);

Man beachte:

• Schema-Typen sind dynamisch, d.h. sie werden im Freispeicher (Heap)

angelegt. Die gesamte Verwaltung (Allozierung, Disallozierung, Zu-

griff über Zeiger etc.) erfolgt automatisiert.

• Gültigkeitsbereich der Objekte ist der aktuelle Block, beginnend bei

der Definition.

• Die Dimensionen werden nur durch die Größe des verfügbaren Frei-

speichers begrenzt.

• Zugriff auf Schemakomponenten erfolgt wie bei arrays über die In-

dizes in [ ]

• Schema-Variablen
”
kennen“ ihre Diskriminanten; Zugriff kann über

die Recordkomponente <name>.<diskriminante> erfolgen.

Beispiele (s.o.):

writeln (x.n, W.maxi); { liefert 450 und 200000 }

readln ( W[123456] ); { Eingabe eines Zeichens }

for i:= 1 to a.m do a[i,i] := 3.5*i;

Variable Diskriminantenwerte

Schema-Typen können in inneren Blöcken (Prozeduren, Funktionen) de-

finiert werden. Dann dürfen die benötigten variablen Diskriminantenwer-

te als Parameter übergeben werden.

Beispiel 8 (s.o.) Skalarprodukt zweier Vektoren

Von gegebenen Vektoren x und y mit n Komponenten ist das Skalar-

produkt s = x · y zu berechnen. Die Größe von n soll beliebig sein

und während des Laufes eingelesen werden!

Program schema1; { Skalarprodukt mit Schema-Typ }

var m : integer;

procedure main (n : integer); { n - Dimension }

type

vektor(n: integer) = array[1..n] of double;

var

x,y : vektor(n); { Schema-Variablen }

s : double;

i : integer;

begin

for i:= 1 to n do

begin

x[i] := 1.0; y[i] := i;

end;

s := 0;

for i:= 1 to n do

s := s + x[i] * y[i];

writeln(’Ergebnis s = ’,s:20:0);

readln

end;

begin

writeln; writeln;

writeln(’Skalarprodukt zweier n-Vektoren’);

writeln(’===============================’);

write(’Dimension n = ’); readln(m);

main(m); { "Hauptprozedur" main }

end.

Skalarprodukt zweier n-Vektoren

===============================

Dimension n = 100000

Ergebnis s = 5000050000

Skalarprodukt zweier n-Vektoren

===============================

Dimension n = 1000000

Ergebnis s = 500000500000

Schema-Typen können mehrdimensional sein. Damit lassen sich großdi-

mensionale Matrizen und Vektoren bequem verarbeiten. Die Größen der

Dimensionen m,n dürfen während der Laufzeit eingelesen werden!

Beispiel 9 Matrix ∗ Vektor

Das Produkt x = A∗b einer (n, n)−MatrixA mit einem n−Vektor

b ist zu bilden und auszugeben. Die Größe von n soll beliebig sein und

während des Laufes eingelesen werden!

Program schema2; { Matrix*Vektor mit Schema-Typen }

var n : integer;

procedure main (m,n : integer);

type

matrix(m,n: integer) = array[1..m,1..n] of double;

vektor(n: integer) = array[1..n] of double;

var

a : matrix(n,n);

b,x : vektor(n);

s : double;

i,k : integer;

procedure Eingabe(var a: matrix; var b:vektor);

var i,k : integer;

begin

for i:= 1 to b.n do begin

for k:=1 to b.n do

a[i,k] := i+k;

b[i] := i*i;

end

end;

procedure Ausgabe(var x: vektor);

var i : integer;

begin

for i:= 1 to x.n do

writeln(’x[’,i,’] = ’, x[i]);

readln

end;

procedure Produkt(var a:matrix; var b:vektor;

var x:vektor);

var

i,k : integer;

s : double;

begin

for i:=1 to b.n do

begin

s := 0;

for k:=1 to b.n do

s := s + a[i,k] * b[k];

x[i] := s

end

end; { of Produkt }

begin { of main }

writeln;

Eingabe(a,b);

Produkt(a,b,x);

{ Ausgabe(x); }

writeln(’Dimension n = ’,x.n); readln

end;

begin { of Program }

writeln; writeln;

writeln(’Matrix * Vektor mit Schema-Typen’);

writeln(’================================’);

write(’Dimension n = ’);

readln(n);

main(n,n);

end.

Der Schema-Typ String

Der Typ String in GPC ist ein vordefinierter (built-in) Schema-Typ mit

der inneren Struktur

String (Capacity : Cardinal) = record

Length : 0 .. Capacity;

Chars : packed array [1 .. Capacity + 1] of Char

end;

GPC-Strings können praktisch beliebige Länge besitzen. Es gilt nicht die

255-Zeichen-Grenze. Man beachte insbesondere:

• Die maximale Länge einer String-Variablen z kann mit z.Capacity

erhalten werden. Die aktuelle Länge von z kann mittels der Funktion

Length(z) abgefragt werden.

• Die BP-Funktionen und -Prozeduren

Length, Pos, Str, Val, Copy, Insert, Delete

sind anwendbar.

• Variablen-Parameter des Types String ohne Diskriminante können

für Schema-Typen benutzt werden. Ansonsten wird die BP-Typver-

einbarung String als String(255) ausgewertet.

Beispiel 10 (s.o.) Textanalyse

Es ist festzustellen, wie oft eine bestimmte Zeichenfolge z der maxima-

len Länge 500 in einem gegebenen Text zk enthalten ist. Dessen Maxi-

mallänge maxi soll beliebig sein und erst während des Laufes eingelesen

werden.

Hinweis: In GPC dürfen Deklarationen auch im Anweisungsteil eines Pro-

grammes stehen.

Program Anzahl1; { String-Schema in GPC }

type

string500 = string(500); { Suchkette }

var

maxi : Integer;

z : string500;

function Anzahl (var zk : string; { Geg. Text }

var z : string500) { Suchkette }

: integer; { Anzahl }

var a, p, l : integer;

begin

l := length(z)-1; { Hilfsgroesse }

a := 0; { Zaehler fuer Anzahl }

p := Pos(z,zk); { 1.Position von z in zk }

while p <> 0 do begin

inc(a);

delete(zk,1,p+l); { Abschneiden }

p := Pos(z,zk) { 1.Position von z in zk }

end; {of while }

Anzahl := a

end; {of Anzahl }

begin

writeln;

repeat

write (’Maximallaenge des Textes : ’);

readln (maxi)

until maxi >= 1;

var zk: string(maxi); { ... in GPC zulaessig! }

write(’Eingabe des Textes zk : ’); readln(zk);

write(’Eingabe der Suchkette z : ’); readln(z);

writeln(’Laenge von zk = ’, length(zk));

writeln(’Laenge von z = ’, length(z));

writeln(’Anzahl des Auftretens : ’,

Anzahl(zk,z));

readln

end.

Hinweis:

GPC ist der freie 32/64-bit Compiler der GNU Compiler-Sammlung.

Er unterstützt (Version 20010409)

• ISO-7185 Standard Pascal

• große Teile von ISO-10206 Extended Pascal

• Borland Pascal 7.0 (mit einigen Delphi-Erweiterungen)

• Teile von Pascal-SC (PXSC).

Vgl. The GNU Pascal Manual, Free Software Foundation. Alle Informa-

tionen und Downloads unter

www.gnu-pascal.de

